


TRƯỜNG ĐẠI HỌC GIAO THÔNG VẬN TẢI
KHOA CÔNG TRÌNH
BỘ MÔN TỰ ĐỘNG HÓA THIẾT KẾ CẦU ĐƯỜNG

Bài Giảng

Tin học xây dựng



NXB Đại học Giao thông vận tải - 2012

Lê Quỳnh Mai (chủ biên) - Trần Thế Hiệp - Bùi Công Độ - Lê Đắc Hiền
Hoàng Thùy Linh - Nguyễn Thị Thanh Yên - Đỗ Xuân Cảnh - Phan Thị Thu Hiền

Bài Giảng

Tin học xây dựng

(DÙNG CHO HỆ CHÍNH QUY – BẢNG 2)

NXB Đại học giao thông vận tải - 2012

MỤC LỤC

MỞ ĐẦU.....	1
1. TỔNG QUAN.....	1
2. EXCEL.....	2
3. AUTOCAD.....	3
4. VBA.....	4
DỰ ÁN I: XÂY DỰNG CHƯƠNG TRÌNH TÍNH ĐẶC TRƯNG HÌNH HỌC CỦA MỘT SỐ LOẠI MẶT CẮT CƠ BẢN HAY DÙNG TRONG KẾT CẤU CẦU.....	5
1. NỘI DUNG DỰ ÁN.....	5
1.1. Mô tả yêu cầu.....	5
1.2. Cấu trúc dự án.....	6
1.3. Nội dung kiến thức VBA liên quan đến dự án.....	7
1.4. Nội dung kiến thức lập trình trên Excel liên quan đến dự án.....	7
1.5. Những lưu ý khác.....	7
2. TRÌNH TỰ THỰC HIỆN DỰ ÁN.....	7
2.1. Tạo file Excel mới và lưu lại thành dạng *.xlsm.....	7
2.2. Chạy VBA IDE và tạo Userform.....	8
2.3. Xây dựng giao diện người sử dụng.....	9
2.4. Viết mã lệnh chi tiết cho chương trình.....	12
2.5. Chạy thử chương trình và kiểm tra kết quả.....	19
3. BÀI TẬP.....	19
3.1. Bài tập 1.....	19
3.2. Bài tập 2.....	20
3.3. Lưu ý khi làm lại dự án mẫu.....	21
DỰ ÁN II: CHƯƠNG TRÌNH LẬP BẢNG TỔNG HỢP VẬT TƯ.....	22
1. NỘI DUNG DỰ ÁN.....	22
1.1. Mô tả yêu cầu.....	22
1.2. Cấu trúc dự án.....	23
1.3. Nội dung kiến thức VBA liên quan đến dự án.....	25
1.4. Những lưu ý khác.....	26
2. TRÌNH TỰ THỰC HIỆN DỰ ÁN.....	26
2.1. Lưu file dữ liệu sang dạng xlsm	26
2.2. Tạo sheet “Tong hop vat tu”.....	27
2.3. Chạy VBA IDE và tạo Module.....	27
2.4. Xây dựng giao diện người sử dụng.....	27

2.5. Xây dựng khối chương trình dạng khung	27
2.6. Viết mã lệnh chi tiết cho chương trình con	28
2.7. Chạy thử chương trình và kiểm tra kết quả.....	36
3. BÀI TẬP	38
3.1. Bài tập 1	38
3.2. Bài tập 2	41
3.3. Bài tập 3	42
DỰ ÁN III: XÂY DỰNG CHƯƠNG TRÌNH NHẬP DỮ LIỆU VÀ TỰ ĐỘNG TẠO BẢN VẼ BỐ TRÍ CHUNG TRỤ CẦU THÂN ĐẶC TRONG AUTOCAD	44
1. NỘI DUNG DỰ ÁN	44
1.1. Mô tả yêu cầu	44
1.2. Kịch bản sử dụng và kết quả chương trình	44
1.3. Cấu trúc dự án	44
1.4. Nội dung kiến thức VBA liên quan đến dự án.....	45
1.5. Những lưu ý khác.....	46
2. TRÌNH TỰ THỰC HIỆN DỰ ÁN	46
2.1. Tạo và lưu bản vẽ mẫu xuất phát	46
2.2. Tạo và lưu dự án VBA	46
2.3. Xây dựng giao diện người sử dụng	46
2.4. Xây dựng khối chương trình dạng khung	47
2.5. Viết mã lệnh chi tiết.....	49
2.6. Chạy thử chương trình và kiểm tra kết quả.....	54
3. BÀI TẬP	54
3.1. Bài tập thực hành	54
3.2. Câu hỏi ôn tập	55
DỰ ÁN IV: CHƯƠNG TRÌNH VẼ TRẮC ĐỌC TỰ NHIÊN TRÊN AUTOCAD VỚI SỐ LIỆU ĐỌC TỪ FILE EXCEL.....	56
1. NỘI DUNG DỰ ÁN	56
1.1. Mô tả yêu cầu	56
1.2. Cấu trúc dự án	57
1.3. Nội dung kiến thức VBA liên quan đến dự án.....	58
1.4. Những lưu ý khác.....	58
2. TRÌNH TỰ THỰC HIỆN DỰ ÁN	59
2.1. Chạy VBA IDE và tạo Module	59
2.2. Lưu trữ dự án dạng DVB	59


2.3. Khai báo tham chiếu hệ thống đối tượng của Excel trong AutoCAD	59
2.4. Xây dựng giao diện người sử dụng	60
2.5. Xây dựng các chương trình con	60
2.6. Chạy thử chương trình và kiểm tra kết quả.....	66
3. BÀI TẬP	67
3.1. Bài tập 1	67
3.2. Bài tập 2	67
DỰ ÁN V: CHƯƠNG TRÌNH CHUYỂN ĐỔI SỐ LIỆU TỪ BẢN VẼ KHẢO SÁT ĐỊA HÌNH TRÊN AUTOCAD SANG SỐ LIỆU DẠNG TỌA ĐỘ ĐIỂM TRÊN EXCEL ĐỂ LÀM DỮ LIỆU CHO CÁC PHẦN MỀM THIẾT KẾ ĐƯỜNG.....	69
1. NỘI DUNG DỰ ÁN	69
1.1. Mô tả yêu cầu.....	69
1.2. Cấu trúc dự án.....	70
1.3. Nội dung kiến thức VBA liên quan đến dự án.....	72
1.4. Những lưu ý khác.....	72
2. TRÌNH TỰ THỰC HIỆN DỰ ÁN	73
2.1. Mở file KS1.dwg.....	73
2.2. Chạy VBA IDE và tạo Module	73
2.3. Xây dựng giao diện người sử dụng	74
2.4. Xây dựng khối chương trình dạng khung	74
2.5. Viết mã lệnh chi tiết cho chương trình chuyển đổi dữ liệu.....	74
2.6. Chạy thử chương trình chuyển đổi dữ liệu	83
2.7. Viết mã lệnh cho chương trình kiểm tra sơ bộ dữ liệu trong Excel	83
3. BÀI TẬP	85
3.1. Bài tập 1	85
3.2. Bài tập 2	86

MỞ ĐẦU


1. TỔNG QUAN

Hiện nay, máy tính đóng một vai trò quan trọng trong quá trình thiết kế cũng như thi công công trình. Với vai trò là một công cụ làm việc, nó có thể giúp kỹ sư thiết kế lập bản vẽ nhanh chóng và chính xác bằng phần mềm AutoCAD, có thể giúp cho việc tính toán mặt cắt và trình bày hồ sơ thiết kế một cách hiệu quả bằng phần mềm Excel, có thể giúp tính toán kết cấu cầu, cống, nhà cửa, cột tháp ... rất nhanh chóng và chính xác bằng phần mềm MIDAS, có thể tính toán kết cấu nền đường, mái dốc bằng phần mềm GeoSlope hoặc Plaxis, có thể giúp thiết kế đường ô tô bằng phần mềm Nova-TDN, ADS-Road hay Civil 3D. Đối với kỹ sư thi công, máy tính có thể hỗ trợ việc quản lý dự án bằng phần mềm Project hay CSM, có thể hỗ trợ tính toán chi phí xây dựng bằng phần mềm dự toán G8 hoặc CE.

Như vậy, để có thể nhanh chóng hoàn thành với chất lượng cao: bài tập lớn, thiết kế môn học, đồ án tốt nghiệp hay dự án thiết kế hoặc thi công thực tế, bạn cần tận dụng sức mạnh của máy tính thông qua việc khai thác các phần mềm thay vì vẽ hoặc tính toán thủ công.



Với sự đa dạng của thực tế thiết kế, thi công hay chính trong các bài tập lớn của các bạn khi đang học, không có một phần mềm nào có thể đáp ứng hay giải quyết được hết mọi vấn đề, do đó, bên trong nhiều phần mềm, ngoài những tính năng có sẵn giúp bạn giải quyết ngay lập tức các vấn đề chính, chúng còn cho phép bạn tự xây dựng những tính năng mới với công sức lập trình ít nhất, hiệu quả cao nhất, để giải quyết các vấn đề phát sinh mà bạn gặp phải trong quá trình làm việc và học tập của cá nhân. Điển hình là phần mềm AutoCAD và Excel, đây là 2 phần mềm được sử dụng phổ biến nhất trong giới kỹ thuật, ngoài các tính năng chính như hỗ trợ lập bản vẽ và tính toán như đã biết, chúng còn tích hợp sẵn công cụ lập trình đơn giản, dễ sử dụng nhưng hiệu quả cao, được biết đến với cái tên đơn giản là **VBA**, nhằm mục đích giúp người dùng có thể tự xây dựng những chương trình, chạy trên nền của chúng, để giải quyết các bài toán hay những vấn đề phát sinh trong thực tế công tác và học tập.



Bạn sẽ băn khoăn trước khi bắt đầu môn học này vì bạn chưa quen việc lập trình cho dù bạn đã học về nó trong môn Tin học đại cương, bạn cũng băn khoăn vì chưa sử dụng nhiều Excel và AutoCAD trong quá trình học tập từ trước đến nay, cho dù chúng đơn giản và bạn đã từng làm quen trong môn Vẽ kỹ thuật. Tất cả những trở ngại này sẽ được giải quyết với một cách học rất thân thuộc với sự học hỏi tự nhiên của con người, nó tương tự như việc bạn học đi xe đạp! Có lẽ không ai muốn tập đi xe đạp mà lại nghiên cứu trước để hiểu rõ vì sao xe đạp 2 bánh lại có thể đứng khi chạy và khi dừng thì không thể đứng, mà đa số mọi người, trong đó có bạn, sẽ không quan tâm đến việc nghiên cứu đó, không quan tâm mấy đến các cơ cấu cơ khí của xe đạp, và thay vào việc nghiên cứu đó, bạn sẽ bắt đầu tập trên xe thực, bắt đầu bị ngã, bắt đầu chưa thể đi được, nhưng sau một thời gian ngắn, bạn đã có một cảm giác mới về khả năng của mình. Việc học lập trình VBA cũng vậy, hãy mạnh dạn bắt đầu với sự hiểu biết không nhiều nhưng với một tinh thần mạnh mẽ, bạn sẽ có được cảm giác thú vị khi kết thúc môn học này.

Mục tiêu chính của quá trình đào tạo được thể hiện ở đề án tốt nghiệp, nơi mà có đến 2/3 kiến thức liên quan đến thiết kế công trình giao thông, do đó, môn học Tin học xây dựng này được thiết kế theo mục tiêu chính này. Hồ sơ thiết kế công trình giao thông, với khối lượng chủ yếu là bản tính và bản vẽ (chiếm hơn 90%), và thực tế tại các công ty đã chứng tỏ, những nội dung này đều được thực hiện trên AutoCAD và Excel.

Với vai trò của kỹ sư thiết kế, bạn không thể chỉ biết đến AutoCAD và Excel ở mức độ cơ bản là biết vẽ hay biết tính toán, mà cần biết ở mức độ chuyên nghiệp hơn, đó là biết điều khiển AutoCAD và Excel thực hiện theo ý tưởng của bạn, và đây chính là mục tiêu của môn học này.

Để đạt được mục tiêu, bạn cần có cách thức thực hiện, ở đây, VBA chính là cách thức để bạn có thể điều khiển AutoCAD và Excel theo ý mình. Chương trình học được thiết kế theo cách chỉ dẫn bạn thực hiện các dự án từ nhỏ đến lớn, từ đơn giản đến phức tạp. Các dự án là những chương trình, mà có thể sau này, bạn sẽ cần đến nó, nhưng quan trọng hơn, nó chính là cách thức để bạn biết cách thực hành và chủ động học tập, giúp bạn đạt được mục tiêu của môn học này.


2. EXCEL

Microsoft Excel là một phần mềm chuyên xử lý bảng tính của hãng phần mềm nổi tiếng Microsoft. Excel thực sự là một công cụ rất mạnh mẽ phục vụ công tác tính toán, lập bảng biểu... Với các bài toán từ đơn giản đến phức tạp, ta đều có thể sử dụng Excel để giải quyết một cách dễ dàng với rất nhiều tính năng sẵn có:

- Khả năng tổ chức dữ liệu mạnh mẽ với hệ thống các ô, vùng dữ liệu, các bảng tính...;
- Khả năng xử lý dữ liệu như truy vấn, lọc, tính toán... với hệ thống rất phong phú các hàm cơ bản cũng như các hàm chức năng chuyên biệt;
- Khả năng lập báo cáo với cách tổ chức bảng biểu và hệ thống biểu đồ tương đối hoàn chỉnh;

- Khả năng in ấn với nhiều lựa chọn khác nhau.

Với cách tổ chức giống như bảng tính thông thường, Excel là một phần mềm bảng tính trực quan và rất dễ sử dụng. Chính bởi điều này khiến cho Excel là một trong những phần mềm không thể thiếu đối với kỹ sư thiết kế.



Trong thực tế, bạn có thể sử dụng Excel để tính toán và trình bày nội dung cho hầu hết các loại bài tập lớn, đồ án tốt nghiệp hay đồ án thiết kế.

3. AUTOCAD

AutoCAD là một phần mềm hỗ trợ tạo bản vẽ kỹ thuật được dùng phổ biến nhất hiện nay. Đây là sản phẩm của hãng Autodesk và được phát triển liên tục trong nhiều năm nay, điều này thể hiện ở việc cập nhật hàng năm của các phiên bản AutoCAD. Hình vẽ trong AutoCAD được tổ chức chủ yếu theo dạng vector và chuẩn lưu trữ dạng DWG được biết đến như là chuẩn lưu trữ hình vẽ dạng vector hiệu quả nhất thế giới. Để tạo sự thuận lợi tối đa cho người dùng, AutoCAD đã được thiết kế với cấu trúc và tính năng rất hợp lý:

- Không gian để tạo bản vẽ được chia thành hai loại:
 - Không gian mô hình (Model), là nơi mà người dùng có thể vẽ hay dựng mô hình của bất cứ vật thể nào mà không cần quan tâm đến giới hạn về kích thước của đối tượng, của bản vẽ cũng như tỷ lệ trình bày.
 - Không gian trình bày hay còn gọi là không gian in (Layout), là nơi mà người dùng có thể vẽ hay dựng mô hình như không gian mô hình, nhưng đây không phải là mục đích chính của không gian in. Mục đích chính của không gian in là giúp cho người dùng có thể biểu diễn hoặc trình bày bản vẽ theo ý tưởng của mình dựa trên mô hình đã được dựng (hay đã được vẽ) trong không gian mô hình. Trong không gian in, với số lượng không hạn chế, người dùng có thể dễ dàng tạo ra những bản in có tỷ lệ khác nhau, cách bố trí, sắp đặt khác nhau từ một mô hình đã vẽ.


- Thao tác tạo bản vẽ được thực hiện thông qua các lệnh trong thanh trình đơn, thanh công cụ, và đặc biệt là thông qua dòng lệnh của AutoCAD. Với hàng trăm lệnh sẵn có, cùng với cách thực hiện lệnh đa dạng, cho nên người dùng có thể làm quen và sử dụng AutoCAD trong một thời gian ngắn.
- Hình vẽ trong AutoCAD, cho dù đơn giản hay phức tạp đến mấy, đều được tạo nên từ những đối tượng hình học cơ bản. Và những đối tượng hình học cơ bản này lại được một hệ thống các đối tượng phi hình học khác trong AutoCAD hỗ trợ việc tạo ra chúng.
- Với cách tổ chức các đối tượng hình học theo lớp (Layer), AutoCAD cho phép người dùng tổ chức bản vẽ, cho dù phức tạp đến mấy, thành từng lớp theo những chủ đề khác nhau, khiến cho việc quản lý và thao tác với bản vẽ trở nên dễ dàng hơn.
- Các tiện ích vẽ in bản vẽ khiến cho việc in ấn trở nên đơn giản và chuyên nghiệp.
- Khi những tính năng sẵn có của AutoCAD không đáp ứng được nhu cầu của người dùng thì người dùng có thể sử dụng khả năng cho phép lập trình mở rộng của AutoCAD để bổ sung thêm hay tạo mới những tính năng chuyên biệt cho AutoCAD nhằm đáp ứng được nhu cầu cá nhân.

4. VBA

VBA (Visual Basic for Applications) là một công cụ lập trình cho phép xây dựng nhanh và hiệu quả chương trình nhằm bổ sung thêm những tính năng mới cho ứng dụng nền (AutoCAD, Excel). Một chương trình được xây dựng bằng VBA dựa trên ứng dụng nền nào thì nó phụ thuộc chặt chẽ vào ứng dụng nền đó, bởi theo mặc định, dự án VBA sẽ hoạt động và sử dụng các thành phần trong chính ứng dụng nền đó. Điều này có nghĩa là ta rất khó có thể chuyển đổi một dự án VBA từ loại ứng dụng nền này sang một ứng dụng nền khác cũng như tạo ra một ứng dụng chạy độc lập.

VBA, thực chất là sự kết hợp giữa công cụ lập trình Visual Basic (VB) và các thành phần được phép sử dụng của ứng dụng nền, cho nên về cơ bản, việc xây dựng một chương trình bằng VBA trên AutoCAD hoặc Excel là giống nhau.

Việc sử dụng VBA trong thực tế diễn ra phổ biến do tính dễ dùng và hiệu quả cao, hiện tại có rất nhiều tài liệu cũng như kiến thức về VBA được chia sẻ trên Internet, do đó, việc lập trình với VBA trở nên thuận tiện hơn.




Trong môn học này, VBA được đề cập đến như một công cụ lập trình trên AutoCAD và Excel, thông qua các dự án mẫu, cùng với các bài tập đi kèm, từng bước hướng dẫn bạn làm quen và làm chủ VBA.




DỰ ÁN I: XÂY DỰNG CHƯƠNG TRÌNH TÍNH ĐẶC TRƯNG HÌNH HỌC CỦA MỘT SỐ LOẠI MẶT CẮT CƠ BẢN HAY DÙNG TRONG KẾT CẤU CẦU

1. NỘI DUNG DỰ ÁN

1.1. Mô tả yêu cầu

- Tên dự án: Xây dựng chương trình tính đặc trưng hình học của một số loại mặt cắt cơ bản hay dùng trong kết cấu cầu.
- Giao diện của chương trình được xây dựng bằng Userform và các điều khiển (Controls) trên VBA IDE của Excel.




- Khi chạy chương trình, người dùng sẽ lựa chọn loại mặt cắt cần tính, ở đây là: mặt cắt chữ nhật, mặt cắt tròn đặc và mặt cắt tròn rỗng. Chương trình sẽ yêu cầu người dùng nhập các kích thước tương ứng với từng loại mặt cắt được chọn.
- Sau khi nhập số liệu, bấm nút lệnh  thì chương trình sẽ thực hiện:
 - Tính diện tích mặt cắt ngang, mômen quán tính của mặt cắt đối với trục y và trục z (các trục chính đi qua trọng tâm của mặt cắt).
 - Ghi kết quả lên Userform (trong mục “Ket qua tinh toan”).
- Khi bấm nút lệnh  thì chương trình sẽ thực hiện:
 - Tính diện tích mặt cắt ngang, mômen quán tính của mặt cắt đối với trục y và trục z (các trục chính đi qua trọng tâm của mặt cắt)
 - Ghi kết quả lên Userform
 - Xuất kết quả vào sheet1 của Excel.
- Khi bấm nút lệnh  thì sẽ thoát ra khỏi chương trình.

1.2. Cấu trúc dự án


1.2.1. Giao diện của chương trình:

- Giao diện của chương trình được xây dựng bằng Userform và điều khiển (Hộp danh sách tổ hợp - ComboBox, Nhãn - Label, Hộp văn bản - TextBox, Khung - Frame, Nút lệnh - CommandButton, Hình ảnh - Image).



1.2.2. Khối chương trình

- Chương trình được đặt trong cửa sổ mã lệnh của Userform.



- Khối mã lệnh định nghĩa các biến được sử dụng trong chương trình.
- Các khối mã lệnh thực thi chức năng của chương trình tương ứng với từng sự kiện trên nút lệnh được bố trí trong các chương trình con.

1.3. Nội dung kiến thức VBA liên quan đến dự án

- Những qui định về cú pháp: mục 1, trang 23.
- Các trợ giúp về cú pháp trong quá trình viết mã lệnh: mục 2, trang 23.
- Các kiểu dữ liệu cơ bản: mục 5, trang 26.
- Khai báo hằng số và khai báo biến: mục 6.1 và 6.2 trang 37.
- Các toán tử và các hàm toán học: mục 7.1 và 7.2, trang 39 và 40,
- Cấu trúc lựa chọn: mục 8.2 trang 45.
- Cách thức gọi chương trình con: mục 9.5, trang 57.
- Làm việc với Userform và các thành phần điều khiển: mục 11 trang 59.

1.4. Nội dung kiến thức lập trình trên Excel liên quan đến dự án

- Đối tượng Workbook: mục 5.2.2 trang 127.
- Đối tượng Worksheet: mục 5.2.4 trang 133.
- Đối tượng Range: mục 5.2.4 trang 135.


1.5. Những lưu ý khác

- Phần hướng dẫn này phù hợp với phiên bản Excel từ 2007 trở lên.
- Tài liệu tham khảo (mục, trang) được lấy theo Giáo trình môn Tự động hóa Thiết kế cầu đường (bản PDF).
- Khi thực hiện cần Save As file sang định dạng *.xlsm, nếu không sử dụng định dạng này thì mã VBA sẽ mất khi đóng Excel.

2. TRÌNH TỰ THỰC HIỆN DỰ ÁN


2.1. Tạo file Excel mới và lưu lại thành dạng *.xlsm

- Mở Excel, tạo file mới.
- Lưu file mới thành dạng *.xlsm: File → Save As → Save as type: Excel Macro-Enabled Workbook → Save




2.2. Chạy VBA IDE và tạo Userform

- Gọi VBA IDE: bấm Alt + F11, hoặc bấm vào biểu tượng như sau:




- Tạo Userform: Chọn Insert → UserForm



2.3. Xây dựng giao diện người sử dụng


- Lựa chọn các điều khiển thích hợp từ hộp công cụ điều khiển (Control Toolbox) để xây dựng giao diện chương trình như yêu cầu.



★ **Kiến thức tham khảo**


✓ Làm việc với Userform và các thành phần điều khiển: mục 11, trang 59.

- Thay đổi các thuộc tính của UserForm và thêm điều khiển Hộp danh sách tổ hợp (ComboBox) bằng cách bấm vào biểu tượng ComboBox trên Toolbox và kéo thả vào UserForm.




- Thêm 3 khung (frame) để nhập dữ liệu tương ứng với 3 loại mặt cắt là chữ nhật, tròn đặc và trong rỗng.


Khung 1 (Frame1): dùng để nhập dữ liệu cho mặt cắt chữ nhật.



Khung 2 (Frame2): dùng để nhập dữ liệu cho mặt cắt tròn đặc.




Khung 3 (Frame3): dùng để nhập dữ liệu cho mặt cắt tròn rỗng.




Để tạo ra 3 file ảnh, có thể làm bằng cách: vẽ các mặt cắt trong AutoCAD, sau đó dùng phần mềm BetterWMF copy sang phần mềm Paint và lưu lại thành file ảnh.

Nhằm mục đích khi chạy chương trình, người dùng chọn tính các mặt cắt khác nhau thì frame nhập số liệu đều ở cùng một vị trí trên Userform. Thực hiện sắp xếp 3 frame (frame1, frame2 và frame3) vào cùng một vị trí trên Userform bằng cách di chuyển 3 frame vào cùng một vị trí hoặc thay đổi các thuộc tính Left, Top của 3 frame.

- Thêm 1 khung (frame) để xuất kết quả tính toán gồm diện tích mặt cắt ngang và mô men quán tính theo hai trục.




- Thêm 3 nút lệnh CommandButton vào Userform:



2.4. Viết mã lệnh chi tiết cho chương trình

- Bấm chuột phải vào Userform, chọn View Code để hiện ra cửa sổ mã lệnh.



★ **Kiến thức tham khảo**

- ✓ Môi trường phát triển tích hợp VBA IDE: mục 4 trang 19.
- ✓ Các trợ giúp về cú pháp trong quá trình viết mã lệnh: mục 2, trang 23.

2.4.1. Định nghĩa các biến chính của chương trình

- Mã lệnh khai báo các biến chính của chương trình:

```

1 Option Explicit
2 ' Khai báo 3 biến: A là diện tích mặt cắt ngang và Iy, Iz là momen quán tính
3 Private A As Double, Iy As Double, Iz As Double
4 ' Khai báo 2 biến B và H là bề rộng và chiều cao mặt cắt chu nhật
5 Dim B As Double, H As Double
6 ' Khai báo biến D là đường kính mặt cắt tròn đặc
7 ' D1 và D2 là đường kính ngoài và đường kính tròn mặt cắt tròn rỗng
8 Dim D As Double, D1 As Double, D2 As Double
9 ' Khai báo hằng số pi
10 Private Const PI = 3.14
    
```

★ **Kiến thức tham khảo**

- ✓ Khai báo biến và hằng số: mục 6.1 và 6.2 trang 37.

2.4.2. Mã lệnh cho thủ tục sự kiện Initialize của Userform

- Mã lệnh:

```
11 Private Sub UserForm_Initialize()  
12     ' Thêm giá trị "Chu nhật" vào trong danh sách của ComboBox, o vị trí đầu tiên (0)  
13     UserForm1.ComboBox1.AddItem "Chu nhật", 0  
14     ' Thêm giá trị "Tròn đặc" vào trong danh sách của ComboBox, o vị trí tiếp theo (1)  
15     UserForm1.ComboBox1.AddItem "Tròn đặc", 1  
16     ' Thêm giá trị "Tròn rỗng" vào trong danh sách của ComboBox, o vị trí tiếp theo (2)  
17     UserForm1.ComboBox1.AddItem "Tròn rỗng", 2  
18     ' Ẩn khung FrmTrondac và FrmTronrong  
19     FrmTrondac.Visible = False  
20     FrmTronrong.Visible = False  
21     ' Hiện thị khung FrmChunhat  
22     FrmChunhat.Visible = True  
23 End Sub
```

- Mã lệnh cho sự kiện Initialize của Userform sẽ được tự động gọi khi chương trình nạp Userform vào bộ nhớ của máy tính.
- Mã lệnh này sẽ thêm các giá trị vào trong danh sách của ComboBox1 và hiển thị khung nhập dữ liệu cho hình chữ nhật, ẩn khung nhập dữ liệu cho hình tròn đặc và hình tròn rỗng (do phần 3 đã thiết lập thuộc tính Text của ComboBox1 là “Chu nhật”).

★ Kiến thức tham khảo

- ✓ Hộp danh sách và hộp danh sách tổ hợp: Trang 72 và 74.
- ✓ Các thuộc tính của Userform và các thành phần điều khiển: mục 11.1.2, trang 63.

2.4.3. Mã lệnh cho thủ tục sự kiện Change của ComboBox

- Mã lệnh:

```

24 Private Sub ComboBox1_Change ()
25     ' Su dung cau truc lua chon de thuc hien lenh tuong ung voi lua chon loai mat cat
26     Select Case ComboBox1.Text
27         ' Neu nguoi dung chon "Chu nhat"
28         Case "Chu nhat"
29             ' An khung nhap du lieu Tron dac
30             FrmTrondac.Visible = False
31             ' An khung nhap du lieu Tron rong
32             FrmTronrong.Visible = False
33             ' Hien thi khung nhap du lieu Chu nhat
34             FrmChunhat.Visible = True
35         ' Neu nguoi dung chon "Tron dac"
36         Case "Tron dac"
37             ' An khung nhap du lieu Tron rong
38             FrmTronrong.Visible = False
39             ' An khung nhap du lieu Chu nhat
40             FrmChunhat.Visible = False
41             ' Hien thi khung nhap du lieu Tron dac
42             FrmTrondac.Visible = True
43         ' Neu nguoi dung chon Tron rong
44         Case "Tron rong"
45             ' An khung nhap du lieu Tron dac
46             FrmTrondac.Visible = False
47             ' An khung nhap du lieu Chu nhat
48             FrmChunhat.Visible = False
49             ' Hien thi khung nhap du lieu Tron rong
50             FrmTronrong.Visible = True
51     End Select
52 End Sub

```

- Mã lệnh cho sự kiện Change của ComboBox sẽ được thực hiện khi người dùng thay đổi loại mặt cắt ngang cần tính hay chính là thay đổi giá trị (Text) của ComboBox.
- Chương trình có 3 khung (frame) để nhập dữ liệu tương ứng với 3 loại mặt cắt là Chu nhat, Tron dac và Tron rong. Khi người dùng chọn loại mặt cắt mong muốn thì khung nhập dữ liệu cho loại mặt cắt đó sẽ được hiển thị còn hai khung còn lại sẽ được ẩn đi.

★ **Kiến thức tham khảo**

- ✓ Cấu trúc lựa chọn: mục 8.2, trang 45.
- ✓ Các điều khiển thông dụng: mục 11.3, trang 69.

2.4.4. Mã lệnh cho thủ tục sự kiện Click của Command Button

- Mã lệnh cho nút lệnh “Tinh”:

```

53 Private Sub CmdTinh_Click()
54     ' Su dung cau truc lua chon de thuc hien lenh tuong ung voi
55     ' loai mat cat duoc lua chon trong ComboBox1
56     Select Case ComboBox1.Text
57         ' Neu nguoi dung chon mat cat "Chu nhat"
58         Case "Chu nhat"
59             ' Lay gia tri nguoi dung nhap trong Hop van ban TxtB va gan vao bien B
60             B = UserForm1.TxtB.Text
61             ' Lay gia tri nguoi dung nhap trong Hop van ban TxtH va gan vao bien H
62             H = UserForm1.TxtH.Text
63             ' Tinh dien tich mat cat ngang va gan vao bien A
64             A = B * H
65             ' Tinh gia tri momen quan tinh doi voi trục y va gan vao bien Iy
66             Iy = B * H ^ 3 / 12
67             ' Tinh gia tri momen quan tinh doi voi trục z va gan vao bien Iz
68             Iz = H * B ^ 3 / 12
69         Case "Tron dac"
70             D = UserForm1.TxtD.Text
71             A = PI * D ^ 2 / 4
72             Iy = PI * D ^ 4 / 64
73             ' Gan gia tri da co cua bien Iy vao bien Iz
74             Iz = Iy
75         Case "Tron rong"
76             D1 = UserForm1.TxtD1.Text
77             D2 = UserForm1.TxtD2.Text
78             A = PI * (D1 ^ 2 - D2 ^ 2) / 4
79             Iy = PI * (D1 ^ 4 - D2 ^ 4) / 64
80             Iz = Iy
81     End Select
82     ' Gan gia tri cua bien A (sau khi da lam tron) vao Hop van ban TxtA
83     UserForm1.TxtA.Text = Round(A, 3)
84     ' Gan gia tri cua bien Iy (sau khi da lam tron) vao Hop van ban TxtIy
85     UserForm1.TxtIy.Text = Round(Iy, 3)
86     ' Gan gia tri cua bien Iz (sau khi da lam tron) vao Hop van ban TxtIz
87     UserForm1.TxtIz.Text = Round(Iz, 3)
88 End Sub

```

- Mã lệnh cho thủ tục Click của nút lệnh CmdTinh sẽ được thực hiện khi người dùng kích chuột vào nút lệnh “Tinh”.
- Sử dụng cấu trúc lựa chọn để tính toán đặc trưng hình học cho các loại mặt cắt ngang tương ứng với lựa chọn của người dùng ở hộp ComboBox
- Tương ứng với từng loại mặt cắt ngang, tiến hành lấy giá trị do người dùng nhập và gán cho các biến đã được định nghĩa, sau đó tính toán các đặc trưng hình học của mặt cắt ngang và gán cho các biến tương ứng.
- Sau khi tính xong sẽ gán giá trị của các biến đã tính được cho các Hộp văn bản (Textbox) tương ứng.

★ **Kiến thức tham khảo**

- ✓ Hộp văn bản (TextBox): trang 69.
- ✓ Các toán tử và hàm toán học: mục 7.1 và 7.1, trang 39 và 40.

- Mã lệnh cho nút lệnh “Tinh và xuất sang Excel”:

```

89 Private Sub CmdExcel_Click()
90     ' Goi thu tuc su kien Click của nút lệnh "Tinh" để thực thi
91     CmdTinh_Click
92     ' Xóa toàn bộ dữ liệu của vùng dữ liệu từ ô A1 đến B5
93     ' của Sheet đầu tiên của Workbook chứa chương trình VBA này
94     ThisWorkbook.Worksheets(1).Range("A1:B5").Clear
95     ' Gán giá trị cho ô A1 và A2
96     ThisWorkbook.Worksheets(1).Range("A1").Value = "Kết quả tính toán đặc trưng hình học mặt cắt ngang"
97     ThisWorkbook.Worksheets(1).Range("A2").Value = "Loại mặt cắt ngang: "
98     ' Gán giá trị cho ô B2 là Loại mặt cắt ngang người dùng chọn
99     ThisWorkbook.Worksheets(1).Range("B2").Value = ComboBox1.Text
100    ThisWorkbook.Worksheets(1).Range("A3").Value = "Diện tích mặt cắt ngang: A(m2) = "
101    ' Gán giá trị diện tích cho ô B3
102    ThisWorkbook.Worksheets(1).Range("B3").Value = Round(A, 3)
103    ThisWorkbook.Worksheets(1).Range("A4").Value = "Momen quán tính của mặt cắt ngang đối với trục Y: Iy(m4) = "
104    ' Gán giá trị mô men quán tính cho ô B4
105    ThisWorkbook.Worksheets(1).Range("B4").Value = Round(Iy, 3)
106    ThisWorkbook.Worksheets(1).Range("A5").Value = "Momen quán tính của mặt cắt ngang đối với trục Z: Iz(m4) = "
107    ThisWorkbook.Worksheets(1).Range("B5").Value = Round(Iz, 3)
108    ' Chọn toàn bộ cột A và B, sau đó điều chỉnh chiều rộng cho phù hợp với dữ liệu của các ô
109    ThisWorkbook.Worksheets(1).Columns("A:B").EntireColumn.AutoFit
110 End Sub

```

- Mã lệnh cho thủ tục Click của nút lệnh CmdExcel sẽ được thực hiện khi người dùng kích chuột vào nút lệnh “Tinh và xuất sang Excel”.
- Trước tiên phải thực hiện việc tính toán để có kết quả đặc trưng hình học (trường hợp người dùng không bấm nút lệnh “Tinh” mà bấm luôn nút lệnh này). Mã lệnh dòng 91 sẽ gọi thủ tục sự kiện CmdTinh_Click để thực thi.
- Sau khi tính xong thì gán các đoạn văn bản và giá trị cho các ô từ A1 đến B5 của Worksheets(1) là sheet đầu tiên của ThisWorkbook (đây là Workbook chứa chương trình VBA này).

★ Kiến thức tham khảo

- ✓ Cách thức gọi chương trình con: mục 9.5, trang 57.
- ✓ Đối tượng Workbook: mục 5.2.2, trang 127.
- ✓ Đối tượng Worksheet: mục 5.2.4, trang 133.
- ✓ Đối tượng Range: mục 5.2.5, trang 135.

- Mã lệnh cho nút lệnh “Thoát”:

```

111 Private Sub CmdThoat_Click()
112     ' An Userform
113     UserForm1.Hide
114 End Sub

```

★ Kiến thức tham khảo

- ✓ Làm việc với UserForm: mục 11.2, trang 67.


2.4.5. Mã lệnh của toàn bộ chương trình

```
1 Option Explicit
2 Private A As Double, Iy As Double, Iz As Double
3 Dim B As Double, H As Double
4 Dim D As Double, D1 As Double, D2 As Double
5 Private Const PI = 3.14
6
7 Private Sub UserForm_Initialize()
8     UserForm1.ComboBox1.AddItem "Chu nhât", 0
9     UserForm1.ComboBox1.AddItem "Tron dac", 1
10    UserForm1.ComboBox1.AddItem "Tron rong", 2
11    FrmTrondac.Visible = False
12    FrmTronrong.Visible = False
13    FrmChunhat.Visible = True
14 End Sub
15
16 Private Sub ComboBox1_Change()
17     Select Case ComboBox1.Text
18         Case "Chu nhât"
19             FrmTrondac.Visible = False
20             FrmTronrong.Visible = False
21             FrmChunhat.Visible = True
22         Case "Tron dac"
23             FrmTronrong.Visible = False
24             FrmChunhat.Visible = False
25             FrmTrondac.Visible = True
26         Case "Tron rong"
27             FrmTrondac.Visible = False
28             FrmChunhat.Visible = False
29             FrmTronrong.Visible = True
30     End Select
31 End Sub
32
33 Private Sub CmdTinh_Click()
34     Select Case ComboBox1.Text
35         Case "Chu nhât"
36             B = UserForm1.TxtB.Text
37             H = UserForm1.TxtH.Text
38             A = B * H
39             Iy = B * H ^ 3 / 12
40             Iz = H * B ^ 3 / 12
```

38	Case "Tron dac"
39	D = UserForm1.TxtD.Text
40	A = PI * D ^ 2 / 4
41	Iy = PI * D ^ 4 / 64
42	Iz = Iy
43	Case "Tron rong"
44	D1 = UserForm1.TxtD1.Text
45	D2 = UserForm1.TxtD2.Text
46	A = PI * (D1 ^ 2 - D2 ^ 2) / 4
47	Iy = PI * (D1 ^ 4 - D2 ^ 4) / 64
48	Iz = Iy
49	End Select
50	UserForm1.TxtA.Text = Round(A, 3)
51	UserForm1.TxtIy.Text = Round(Iy, 3)
52	UserForm1.TxtIz.Text = Round(Iz, 3)
53	End Sub
54	Private Sub CmdExcel_Click()
55	CmdTinh_Click
56	ThisWorkbook.Worksheets(1).Range("A1:B5").Clear
57	ThisWorkbook.Worksheets(1).Range("A1").Value = "Ket qua tinh toan dac trung hinh hoc mat cat ngang"
58	ThisWorkbook.Worksheets(1).Range("A2").Value = "Loai mat cat ngang: "
59	ThisWorkbook.Worksheets(1).Range("B2").Value = ComboBox1.Text
60	ThisWorkbook.Worksheets(1).Range("A3").Value = "Dien tich mat cat ngang: A(m2) = "
61	ThisWorkbook.Worksheets(1).Range("B3").Value = Round(A, 3)
62	ThisWorkbook.Worksheets(1).Range("A4").Value = "Momen quan tinh cua mat cat ngang doi voi truc Y: Iy(m4) = "
63	ThisWorkbook.Worksheets(1).Range("B4").Value = Round(Iy, 3)
64	ThisWorkbook.Worksheets(1).Range("A5").Value = "Momen quan tinh cua mat cat ngang doi voi truc Z: Iz(m4) = "
65	ThisWorkbook.Worksheets(1).Range("B5").Value = Round(Iz, 3)
66	ThisWorkbook.Worksheets(1).Columns("A:B").EntireColumn.AutoFit
67	End Sub
68	Private Sub CmdThoat_Click()
69	UserForm1.Hide
70	End Sub

2.5. Chạy thử chương trình và kiểm tra kết quả

- Trong VBA IDE: bấm F5 hoặc vào menu Run → Run Sub/Userform



- Chọn loại mặt cắt cần tính, nhập số liệu, bấm nút lệnh để kiểm tra chức năng của chương trình.

3. BÀI TẬP

3.1. Bài tập 1

- Nên làm lại dự án mẫu trước khi làm các bài tập.
- Nội dung: Lập chương trình tính ứng suất lớn nhất trong bài toán uốn và kéo nén đồng thời cho mặt cắt hình chữ nhật và hình tròn đặc bằng VBA trên Excel.
- Hướng dẫn:
 - Dựa trên Dự án mẫu VBA1, phát triển thêm tính năng tính toán ứng suất lớn nhất về kéo và nén của mặt cắt và so sánh với ứng suất cho phép.
 - Số liệu nhập thêm: lực dọc trục N_x , mô men uốn M_y và M_z và ứng suất cho phép chịu kéo và ứng suất cho phép chịu nén.
 - Giao diện của chương trình như sau:


Tính đặc trưng hình học và Ứng suất trên mặt cắt

Chu nhật

Nhập kích thước mặt cắt Tròn rỗng

D1 (m) =

D2 (m) =



Nhập Nội lực và Ứng suất cho phép

N (kN) US kéo cho phép (kN/m²)

My (kN.m) US nén cho phép (kN/m²)

Mz (kN.m)

Kết quả tính toán

A (m²) =

Iy (m⁴) =

Iz (m⁴) =

US kéo lớn nhất (kN/m²)

US nén lớn nhất (kN/m²)

Kết quả so với US cho phép OK/Not OK?

Tính Tính và xuất sang Excel Thoát

- Yêu cầu:
 - Tính ra được ứng suất kéo lớn nhất và ứng suất nén lớn nhất. So sánh với ứng suất cho phép.
 - Xuất kết quả lên UserForm và Excel.

3.2. Bài tập 2

- Nội dung: Lập chương trình giải phương trình bậc 2 bằng VBA trên Excel.
- Hướng dẫn:
 - Xây dựng một Userform và các điều khiển cần thiết.
 - Người dùng nhập giá trị a, b và c.

- Yêu cầu:

- Tính và xuất kết quả (nghiệm x_1 , x_2 hoặc thông báo vô nghiệm) lên UserForm và Excel.

3.3. Lưu ý khi làm lại dự án mẫu

- Hình vẽ minh họa các mặt cắt có sẵn trong thư mục của dự án mẫu, do đó có thể sử dụng ngay khi thiết kế UserForm.
- Do việc sử dụng tiếng Việt có dấu chưa được hỗ trợ tốt trong VBA IDE nên chỉ cần sử dụng tiếng Việt không có dấu là được.
- Kích thước của các thành phần trên UserForm có thể điều chỉnh tùy ý, miễn là đủ để hiển thị các thông tin cần thiết.

DỰ ÁN II: CHƯƠNG TRÌNH LẬP BẢNG TỔNG HỢP VẬT TƯ

1. NỘI DUNG DỰ ÁN

1.1. Mô tả yêu cầu

- Tên dự án: Chương trình lập bảng tổng hợp vật tư
- Dữ liệu ban đầu: bảng phân tích vật tư (như hình dưới) là bảng kê chi tiết các loại vật liệu cần sử dụng để làm một công việc (tên vật liệu, mã vật liệu và khối lượng cần sử dụng). Bảng này giúp người thi công chuẩn bị vật liệu để làm công việc cụ thể. Tuy nhiên đối với người làm cung ứng vật liệu thì họ cần biết loại vật liệu và tổng khối lượng là bao nhiêu. Do đó, bảng tổng hợp vật tư cần phải được lập để đáp ứng yêu cầu này.

BẢNG PHÂN TÍCH VẬT TƯ					
CÔNG TRÌNH: CẢI TẠO, NÂNG CẤP ĐƯỜNG AAA					
HẠNG MỤC : THOÁT NƯỚC					
STT	MÃ HIỆU	MSVT	TÊN CÔNG VIỆC HAO PHÍ VẬT TƯ	ĐƠN VỊ	KHỐI LƯỢNG
1	AK.98120		Làm lớp đá đệm móng tường chắn	m ³	150.00
		:A24.0026	Đá cấp phối D<=6cm	m ³	180.00
		:A24.0180	Cát vàng	m ³	45.00
2	AE.26314		Xây gạch chi tường chắn bằng vữa xi măng mác 75	m ³	258.00
		:A24.0383	Gạch chỉ 6,5x10,5x22	viên	141,900.00
		:A24.0797	Xi măng PC30	kg	27,495.06
		:A24.0176	Cát mịn ML=1,5-2,0	m ³	93.65
		:A24.0524	Nước	lít	22,337.64
3	AE.11215		Xây đá hộc cửa cống, vữa xi măng mác 100	m ³	168.00
		:A24.0031	Đá hộc	m ³	201.60
		:A24.0010	Đá 4x6	m ³	9.58
		:A24.0797	Xi măng PC30	kg	27,168.42
		:A24.0180	Cát vàng	m ³	76.91
		:A24.0524	Nước	lít	18,345.60
4	AK.98120		Làm lớp đá dăm đệm móng cống	m ³	213.36
		:A24.0026	Đá cấp phối D<=6cm	m ³	256.03
		:A24.0180	Cát vàng	m ³	64.01
5	AE.26314		Xây gạch chi tường rãnh bằng vữa xi măng mác 75	m ³	547.06
		:A24.0383	Gạch chỉ 6,5x10,5x22	viên	300,880.80
		:A24.0797	Xi măng PC30	kg	58,299.75
		:A24.0176	Cát mịn ML=1,5-2,0	m ³	198.56
		:A24.0524	Nước	lít	47,364.11
6	AF.11111		Bê tông lót móng tường chắn mác 100	m ³	313.02
		:A24.0797	Xi măng PC30	kg	62,870.67
		:A24.0180	Cát vàng	m ³	166.37
		:A24.0010	Đá 4x6	m ³	293.07
		:A24.0524	Nước	lít	53,198.26
7	AG.11414		Bê tông tấm đan nắp rãnh, mác 250	m ³	152.21
		:A24.0797	Xi măng PC30	kg	62,570.55
		:A24.0180	Cát vàng	m ³	68.60
		:A24.0008	Đá 1x2	m ³	133.64
		:A24.0524	Nước	lít	28,581.61

- Dữ liệu có trong file “Phan tich vat tu.xlsx”, thực hiện lập trình bắt đầu với file này.
- Bảng tổng hợp vật tư bao gồm danh mục các vật liệu có trong bảng phân tích vật tư (mỗi loại vật liệu được ghi trên 1 dòng) và tổng khối lượng cần có (như hình dưới)

BẢNG TỔNG HỢP VẬT TƯ				
CÔNG TRÌNH: CAI TẠO, NÂNG CẤP ĐƯỜNG AAA				
HẠNG MỤC : THOÁT NƯỚC				
TT	MSVT	VẬT TƯ HAO PHI	ĐV	KHỐI LƯỢNG
1	A24.0026	Đá cấp phối D<=6cm	m3	455.25
2	A24.0180	Cát vàng	m3	579.20
3	A24.0383	Gạch chỉ 6.5x10.5x22	viên	456881
4	A24.0797	Xi măng PC30	kg	367398.85
5	A24.0176	Cát mịn ML=1,5-2,0	m3	301.52
6	A24.0524	Nước	lít	246655.39
7	A24.0031	Đá hộc	m3	236.83
8	A24.0010	Đá 4x6	m3	316.34
9	A24.0008	Đá 1x2	m3	364.47
10	A24.0738	Thép tròn D<=10mm	kg	22775.31
11	A24.0293	Dây thép	kg	485.42
12	A24.0418	Gỗ ván	m3	15.58
13	A24.0405	Gỗ đũa, chông	m3	11.35
14	A24.0054	Đinh	kg	74.69
15	A24.0056	Đinh đĩa	cái	265
16	A24.0420	Gỗ ván (cả nẹp)	m3	0.31
17	A24.0175	Cát mịn ML=0,7-1,4	m3	73.41
18	A24.0226	Cọc tre 2,5m	m	2962.47
19	A24.0183	Cây chông	cây	44

- Chương trình sẽ yêu cầu chọn vùng dữ liệu chứa các thông tin về các loại vật liệu của tất cả các công việc (vùng dữ liệu **C6:F116** trong sheet “Phan tích vật tu” của file “Phan tích vật tu.xlsx”).
- Chương trình tự động xác định danh mục các loại vật liệu có trong sheet “Phan tích vật tu” và tính tổng khối lượng của từng loại vật liệu tương ứng. Sau đó ghi kết quả vào sheet “Tong hop vật tu”.

1.2. Cấu trúc dự án


1.2.1. Dữ liệu ban đầu

- File “Phan tích vật tu.xlsx”: chứa dữ liệu ở sheet “Phan tích vật tu”, các sheets khác không có dữ liệu.
- Định dạng file này theo chuẩn của Excel 2007 (hoặc mới hơn).




1.2.2. Giao diện người dùng

- Chương trình được sử dụng dưới dạng Macro (xem mục 2.3 trang 108).




- Người dùng chọn vùng dữ liệu cần tổng hợp

STT	MSVT	TÊN CÔNG VIỆC HAO PHÍ VẬT TƯ	ĐƠN VỊ	KHỐI LƯỢNG
20		Làm lớp đá đệm móng tường chân	m3	150.00
	A24.0026	Đá cấp phối D<=6cm	m3	180.00
	A24.0180	Cát vàng		
14		Xây gạch chỉ tường chân bản mức 75		
	A24.0383	Gạch chỉ 6,5x10,5x22		
	A24.0797	Xi măng PC30		
	A24.0176	Cát mịn ML=1,5-2,0		
	A24.0524	Nước		
15		Xây đá học cửa công, vữa xi măng mức 100	m3	168.00
	A24.0031	Đá học	m3	201.60
	A24.0010	Đá 4x6	m3	9.58



1.2.3. Khối chương trình (Module)

- Chương trình được đặt trong 01 Module.



- Khối mã lệnh định nghĩa kiểu dữ liệu người dùng.
- Khối mã lệnh thực thi chức năng của chương trình được bố trí trong 1 chương trình con.

1.3. Nội dung kiến thức VBA liên quan đến dự án

- Yêu cầu khai báo biến trước khi sử dụng: mục 2 trang 23.
- Định nghĩa và sử dụng kiểu dữ liệu người dùng: mục 5.8 trang 31.
- Chương trình con: mục 9 trang 50 và mục 3.2.2 trang 113
- Khai báo biến cơ bản: mục 5 trang 26 và mục 6 trang 33.
- Khai báo và sử dụng mảng động: mục 5.4 trang 28.
- Cấu trúc **For Each ... In ...** : mục 8.3.2 trang 48.
- Cấu trúc **If ... Then ... Else ... End If** : mục 8.1 trang 42.
- Cấu trúc **For ... To ...** : mục 8.3.1 trang 46.
- Dữ liệu kiểu **Range**: dùng khi cần lưu trữ dữ liệu của 1 hoặc nhiều Cells của Excel (mục 5.2.5 trang 135).
- **Application.InputBox**: dùng để tạo ra giao diện nhập dữ liệu đơn giản cho người dùng khi cần chọn một vùng các Cells trong Excel (mục 8.2.1 trang 165).
- **Trim**: hàm xử lý chuỗi, mục đích là loại bỏ những ký tự trắng (Space – không nhìn thấy) ở phía trước và sau của một chuỗi, ví dụ: “**ABCD**” sẽ khác “ **ABCD** ” (mục 7.4 trang 41).
- Đọc và ghi dữ liệu với **Columns(), Cells()**: mục 5.2.5 trang 135
- **Lbound** và **Ubound**: hàm xác định chỉ số nhỏ nhất và lớn nhất của mảng (mục 9.3.5 trang 54).
- Ghi dữ liệu vào Excel theo địa chỉ thay đổi theo biến: **Cells(row, col)** (mục 5.2.6 trang 140).

- Đọc dữ liệu từ Excel với **Offset(row, col)**: mục 5.2.5 trang 138.


1.4. Những lưu ý khác

- Phần hướng dẫn này phù hợp với phiên bản Excel từ 2007 trở lên.
- Tài liệu tham khảo (mục, trang) được lấy theo Giáo trình môn Tự động hóa Thiết kế cầu đường (bản PDF).
- Khi thực hiện cần **Save As** file sang định dạng *.**xlsm**, nếu không sử dụng định dạng này thì mã VBA sẽ mất khi đóng Excel.


2. TRÌNH TỰ THỰC HIỆN DỰ ÁN

2.1. Lưu file dữ liệu sang dạng **xlsm**

- File dữ liệu ban đầu (Phan tích vật tu.xlsx) chỉ có dữ liệu phân tích vật tư cho từng công việc của công trình.
- Mở file dữ liệu này bằng Excel 2007 hoặc 2010 → File → Save As → Save as type: Excel Macro-Enabled Workbook → Save




2.2. Tạo sheet “Tong hop vat tu”



2.3. Chạy VBA IDE và tạo Module

- Gọi VBA IDE: bấm **Alt + F11**
- Tạo Module: Chọn Insert → Module



2.4. Xây dựng giao diện người sử dụng

- Chương trình này không có giao diện dạng UserForm mà thực hiện thông qua cách gọi Macro của Excel (xem mục 2.3 trang 108).

2.5. Xây dựng khối chương trình dạng khung

Thêm một Module vào dự án với tên được đặt là **Chuongtrinh**.

2.5.1. Viết mã lệnh định nghĩa kiểu dữ liệu người dùng cho **Loại vật tư bất kỳ**

TT	Ý nghĩa	Tên biến	Kiểu dữ liệu	Ví dụ
1	Mã số của vật liệu (vật tư)	MaSo	String	:A24.0026
2	Tên của vật liệu	Ten	String	Đá cấp phối D<=6cm
3	Đơn vị đo đại diện cho loại vật liệu	DonVi	String	m3
4	Khối lượng đại diện cho loại vật liệu	KhoiLuong	Double	180.00

★ **Kiến thức tham khảo**

- ✓ Cách đặt tên biến: mục 1 trang 23
- ✓ Kiểu số thực: mục 5.3 trang 27.

- ✓ Kiểu chuỗi: mục 5.5 trang 29.
- ✓ Khai báo và sử dụng kiểu dữ liệu tự định nghĩa: mục 5.8, trang 31.

2.5.2. Tạo chương trình con trong Module

- Chương trình con này sẽ thực hiện tất cả các nội dung của dự án và là dạng Macro của Excel.

```

10 Public Sub DanhSachVatTu()
11
12 End Sub


```

★ Kiến thức tham khảo

- ✓ Từ khóa **Public**: mục 9 trang 50 (tham khảo thêm mục 6 trang 33).
- ✓ Chương trình con dạng thủ tục (**Sub**): mục 9.2, trang 51.

2.6. Viết mã lệnh chi tiết cho chương trình con

- Bấm đúp vào **Module1**.



★ Kiến thức tham khảo

- ✓ Làm việc với VBA IDE: mục 4 trang 19.

2.6.1. Định nghĩa các biến chính của chương trình

- Mã lệnh khai báo các biến chính của chương trình:

```

10 Public Sub DanhSachVatTu()
11     Dim R As Range ' Phạm vi trong bảng vật liệu cần phân tích vật tư
12     Dim DanhSachVT() As LoaiVatTu ' Mảng dòng chứa danh sách vật tư
13     Dim i As Long ' Chỉ số mảng dòng
14     Dim k As Range ' Biến này dùng để duyệt bảng dữ liệu trong R
15 
```

★ Kiến thức tham khảo

- ✓ Khai báo biến cơ bản: mục 6.2 trang 37.
- ✓ Khai báo biến kiểu tự định nghĩa: mục 5.8 trang 31 và mục 6.3 trang 37.
- ✓ Khai báo mảng động: mục 6.5 trang 38
- ✓ Dữ liệu kiểu **Range**: mục 7.4.1 trang 156

2.6.2. Giao diện nhập dữ liệu chọn một vùng trong Excel

- Mã lệnh:

```

15
16     Set R = Application.InputBox("Chọn vùng dữ liệu cần tổng hợp", Type:=8)
17 
```

★ Kiến thức tham khảo

- ✓ Từ khóa **Set**: mục 6.6 trang 38
- ✓ Đối tượng Application: mục 5.1 trang 121
- ✓ Hàm InputBox: mục 8.2.1 trang 165

2.6.3. Khai báo biến phụ dùng cho vòng lặp

```

17
18     i = 0 ' chỉ số đầu tiên của mảng vật tư là 0
19
20     Dim ii As Long
21     Dim ok As Boolean
22 
```

★ Kiến thức tham khảo

- ✓ Chỉ số của mảng: mục 5.4 trang 28

2.6.4. Duyệt các Cells chứa mã của từng loại vật liệu

- Mã lệnh:

```

22
23     'Đọc dữ liệu từ sheet "Phân tích vật tư"
24     For Each k In R.Columns(1).Cells
25         .....
26     Next

```

- Biến R (kiểu Range) đã chứa toàn bộ vùng dữ liệu (các Cells) của bảng phân tích vật tư được chọn trong mục 6.2.

- Cấu trúc For Each trên sẽ duyệt lần lượt các Cells trong cột số 1 của vùng dữ liệu được chọn, do đó khi chọn dữ liệu cần phải đảm bảo cột đầu tiên được chọn là cột mã số của vật liệu (cột C trong file dữ liệu).
- Vòng lặp For Each trên sẽ duyệt từ Cell đầu tiên cho đến Cell cuối cùng của cột số 1, mỗi lần chạy nó sẽ gán giá trị đọc được vào biến K, tức là biến K sẽ chứa mã số vật liệu hoặc rỗng (do trong cột C này có những Cells rỗng).

	A	B	C	
1	BẢNG PHÂN TÍCH			
2	CÔNG TRÌNH: CẢI TẠO, NÂNG			
3	HẠNG MỤC : THO			
4	Có giá trị			
5	STT	MÃ HIỆU	MSVT	TÊN CÔNG HAO PH
6	1	AK.98120		Làm lớp đá đệm móng
7			A24.0026	Đá cấp phối D<=6cm
8			A24.0180	Cát vàng
9	2	AE.26314		Xây gạch chỉ tường chiều mác 75
10			A24.0383	Gạch chỉ 6,5x10,5x
11			A24.0797	Xi măng PC30
12			A24.0176	Cát mịn ML=1,5-2,6

★ Kiến thức tham khảo

- ✓ Sử dụng For Each và biến kiểu Range: mục 7.4.2 trang 156
- ✓ Cấu trúc lặp xác định: mục 8.3 trang 46

2.6.5. Lọc các Cells có mã số vật liệu (If ... Then)

- Mã lệnh:

```

24 For Each k In R.Columns(1).Cells
25     If Trim(k.Value) <> "" Then
26         ...
27     End If
28 Next
    
```

- Chỉ những dòng có mã vật tư thì mới có dữ liệu, do đó, nếu gặp Cell nào có dữ liệu thì mới thực hiện, còn không có (rỗng) thì bỏ qua và For Each sẽ thực hiện vòng lặp tiếp theo.
- Khi viết mã lệnh cho các cấu trúc điều khiển (xem mục 8 trang 42) thì nên viết hoàn chỉnh khung của cấu trúc ngay để tránh nhầm lẫn (như trên thì từ khóa End If được viết ngay để hoàn chỉnh khung của cấu trúc điều khiển If Then).
- Nên sử dụng cách trình bày thẳng theo cột khi viết mã lệnh cho các cấu trúc, các lệnh bên trong cấu trúc nên được viết lùi vào 1 tab so với cột bắt đầu của cấu trúc.
- Dữ liệu trong các Cells của cột mã số có thể chứa các khoảng trắng phía trước và sau ký hiệu mã lệnh (ký tự Space không nhìn thấy nhưng vẫn tồn tại), do đó cần sử dụng hàm Trim để loại bỏ các khoảng trắng này.

★ Kiến thức tham khảo

- ✓ Sử dụng If ... Then: mục 8.1 trang 42
- ✓ Toán tử so sánh: mục 7.1 trang 39

2.6.6. Lập danh sách vật tư tổng hợp

- Danh sách vật tư tổng hợp chứa trong mảng động đã khai báo ở dòng lệnh 12 dưới dạng một mảng động (mảng chưa biết có bao nhiêu phần tử, chỉ biết tên và kiểu dữ liệu khi khai báo biến).
- Chỉ số đầu tiên của danh sách là 0 (gán ở dòng lệnh 18).
- Khi thực hiện lệnh For Each, mỗi vòng lặp giá trị của i (chỉ số mảng động) sẽ thay đổi. Có 2 trường hợp xảy ra:
 - Nếu $i=0$: mảng động chứa danh sách vật liệu chưa có dữ liệu, do đó dữ liệu đầu tiên đọc được từ vùng R sẽ được lấy làm phần tử đầu tiên trong danh sách.
 - Nếu $i>0$ (ngược lại của trường hợp trên): mảng động chứa danh sách đã có số liệu, cho nên xảy ra 2 tình huống:
 - Nếu mã số vật liệu vừa đọc được (chứa trong K) chưa có trong danh sách thì thêm nó vào danh sách
 - Nếu mã số vật liệu vừa đọc được (chứa trong K) đã có trong danh sách thì cộng dồn khối lượng lại cho loại vật liệu đó
- Mã lệnh chung: (nếu $i=0$ thì ... ngược lại thì...)

```

24 For Each k In R.Columns(1).Cells
25     If Trim(k.Value) <> "" Then
26         If i = 0 Then ' Vat tu dau tien trong danh sach
27             '
28         Else ' neu danh sach vat tu lon hon 1
29             '
30         End If
31     End If
32 Next

```

- Nếu $i=0$ (trong danh sách chưa có vật liệu nào):
 - Cấp phát phần tử đầu tiên cho mảng động (chỉ số đầu và cuối đều là 0 nhưng số phần tử là 1)
 - Gán giá trị đọc được
 - Tăng biến kiểm soát chỉ số lên 1

```

26 If i = 0 Then ' Vat tu dau tien trong danh sach
27     ReDim Preserve DanhSachVT(i)
28     'Gan du lieu cho vat tu dau tien
29     DanhSachVT(i).MaSo = Trim(k.Value)
30     DanhSachVT(i).Ten = Trim(k.Offset(0, 1).Value)
31     DanhSachVT(i).DonVi = Trim(k.Offset(0, 2).Value)
32     DanhSachVT(i).KhoiLuong = k.Offset(0, 3).Value
33     i = i + 1 ' tang chi so mang len 1
34 Else ' neu danh sach vat tu lon hon 1

```

- ✓ Đọc dữ liệu từ Excel với Offset: mục **Offset** trang 138
- ✓ Đọc dữ liệu và gán giá trị cho Cells của Excel với Value: mục **Value** trang 139

- Nếu $i > 0$ (trong danh sách đã có ít nhất 1 loại vật liệu):
 - Duyệt từ đầu danh sách (sử dụng `For ii=0 To i-1`) để kiểm tra xem loại vật liệu nào có trong danh sách có mã số trùng với mã số trong `k.Value`
 - Nếu giá trị trong K (`k.Value` chứa mã số vật liệu) **đã có** trong danh sách thì:
 - Đặt biến `OK = False`
 - Tính tổng khối lượng cho loại vật liệu

```

35      ok = True
36      For ii = 0 To i - 1
37          ' Vat tu nay da co trong danh sach
38          If DanhSachVT(ii).MaSo = Trim(k.Value) Then
39              ok = False
40              DanhSachVT(ii).KhoiLuong = DanhSachVT(ii).KhoiLuong + k.Offset(0, 3).Value
41          Exit For
42      End If
43      Next ii
    
```

- Thoát khỏi vòng `For ii=0 To i-1`

- Nếu giá trị trong K (`k.Value` chứa mã số vật liệu) **chưa có** trong danh sách (khi chạy hết vòng `For ii=0` mà biến `OK` có giá trị là `True`):

```

44          ' Vat tu chua co ten trong danh sach
45          If ok Then
46              ReDim Preserve DanhSachVT(i)
47              DanhSachVT(i).MaSo = Trim(k.Value)
48              DanhSachVT(i).Ten = Trim(k.Offset(0, 1).Value)
49              DanhSachVT(i).DonVi = Trim(k.Offset(0, 2).Value)
50              DanhSachVT(i).KhoiLuong = k.Offset(0, 3).Value
51              i = i + 1
52          End If
    
```

```

53      End If
54  End If
55  Next
    
```

2.6.7. Ghi kết quả vào Excel

- Ghi toàn bộ dữ liệu trong danh sách vật tư (chứa trong mảng động `DanhSachVT`) vào sheet “Tong hop vat tu”.

```

56      'Ghi ket qua ra Excel, trong sheet "Tong hop vat tu"
57      Dim j As Long
58      Dim row As Long
59
60      row = 1 ' Bat dau ghi du lieu tu dong so 1
61      For j = LBound(DanhSachVT) To UBound(DanhSachVT)
62          ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 1).Value = j + 1
63          ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 2).Value = DanhSachVT(j).MaSo
64          ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 3).Value = DanhSachVT(j).Ten
65          ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 4).Value = DanhSachVT(j).DonVi
66          ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 5).Value = DanhSachVT(j).KhoiLuong
67      Next j
68      MsgBox "Ket thuc"
69  End Sub
    
```

- ✓ Ghi dữ liệu vào Excel với `Cells(row, col)`: mục 5.2.6 trang 140 và mục 7.4.2 trang 156
- ✓ Đọc dữ liệu và gán giá trị cho `Cells` của Excel với `Value`: mục `Value` trang 139

2.6.8. Thông báo chương trình chạy xong

```
68 MsgBox "Ket thuc"  
69 End Sub
```

2.6.9. Mã lệnh của toàn bộ chương trình

```

1 Option Explicit
2
3 Type LoaiVatTu
4 MaSo As String
5 Ten As String
6 DonVi As String
7 KhoiLuong As Double
8 End Type
9
10 Public Sub DanhSachVatTu()
11     Dim R As Range ' Phạm vi trong bảng vật liệu cần phân tích vật tu
12     Dim DanhSachVT() As LoaiVatTu ' Mạng dòng chứa danh sách vật tu
13     Dim i As Long ' Chỉ số mạng dòng
14     Dim k As Range ' Bên này dùng để duyệt bảng dữ liệu trong R
15
16     Set R = Application.InputBox("Chọn vùng dữ liệu cần tổng hợp", Type:=8)
17
18     i = 0 ' chỉ số đầu tiên của mạng vật tu là 0
19
20     Dim ii As Long
21     Dim ok As Boolean
22
23     'Doc dữ liệu từ sheet "Phân tích vật tu"
24     For Each k In R.Columns(1).Cells
25         If Trim(k.Value) <> "" Then
26             If i = 0 Then ' Vật tu đầu tiên trong danh sách
27                 ReDim Preserve DanhSachVT(i)
28                 'Gán dữ liệu cho vật tu đầu tiên
29                 DanhSachVT(i).MaSo = Trim(k.Value)
30                 DanhSachVT(i).Ten = Trim(k.Offset(0, 1).Value)
31                 DanhSachVT(i).DonVi = Trim(k.Offset(0, 2).Value)
32                 DanhSachVT(i).KhoiLuong = k.Offset(0, 3).Value
33                 i = i + 1 ' tăng chỉ số mạng lên 1
34             Else ' nếu danh sách vật tu lớn hơn 1
35                 ok = True
36                 For ii = 0 To i - 1
37                     ' Vật tu này đã có trong danh sách
38                     If DanhSachVT(ii).MaSo = Trim(k.Value) Then
39                         ok = False
40                         DanhSachVT(ii).KhoiLuong = DanhSachVT(ii).KhoiLuong + k.Offset(0, 3).Value
41                     Exit For


```

```
42         End If
43     Next ii
44     ' Vat tu chua co ten trong danh sach
45     If ok Then
46         ReDim Preserve DanhSachVT(i)
47         DanhSachVT(i).MaSo = Trim(k.Value)
48         DanhSachVT(i).Ten = Trim(k.Offset(0, 1).Value)
49         DanhSachVT(i).DonVi = Trim(k.Offset(0, 2).Value)
50         DanhSachVT(i).KhoiLuong = k.Offset(0, 3).Value
51         i = i + 1
52     End If
53 End If
54 End If
55 Next
56 'Ghi ket qua ra Excel, trong sheet "Tong hop vat tu"
57 Dim j As Long
58 Dim row As Long
59
60 row = 1 ' Bat dau ghi du lieu tu dong so 1
61 For j = LBound(DanhSachVT) To UBound(DanhSachVT)
62     ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 1).Value = j + 1
63     ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 2).Value = DanhSachVT(j).MaSo
64     ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 3).Value = DanhSachVT(j).Ten
65     ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 4).Value = DanhSachVT(j).DonVi
66     ThisWorkbook.Worksheets("Tong hop vat tu").Cells(row + j, 5).Value = DanhSachVT(j).KhoiLuong
67 Next j
68 MsgBox "Ket thuc"
69 End Sub
```


2.7. Chạy thử chương trình và kiểm tra kết quả

2.7.1. Thực hiện thông qua Macro

- Trong Sheet “Phan tich vat tu” → Bấm **Alt + F8** → Chọn “**DanhSachVT**” trong **Macro name** → Chọn **Run**



- Chọn vùng dữ liệu, yêu cầu vùng này phải bắt đầu từ cột C và kết thúc ở cột F, dòng bắt đầu có thể là bất cứ dòng nào miễn là lớn hơn 5. Nếu chọn toàn bộ vùng dữ liệu (**C6:F116**) thì kết quả như sau:


	A	B ↓	C	D	E
1	1	:A24.0026	Đá cấp p	m3	455.248
2	2	:A24.0180	Cát vàng	m3	579.201
3	3	:A24.0383	Gạch chỉ viên		456881
4	4	:A24.0797	Xi măng	F kg	367399
5	5	:A24.0176	Cát mịn	N m3	301.517
6	6	:A24.0524	Nước	lít	246655
7	7	:A24.0031	Đá hộc	m3	236.827
8	8	:A24.0010	Đá 4x6	m3	316.338
9	9	:A24.0008	Đá 1x2	m3	364.471
10	10	:A24.0738	Thép tròn	kg	22775.3
11	11	:A24.0293	Dây thép	kg	485.42
12	12	:A24.0418	Gỗ ván	m3	15.5768
13	13	:A24.0405	Gỗ đà, c	m3	11.3548
14	14	:A24.0054	Đinh	kg	74.6867
15	15	:A24.0056	Đinh đĩa	cái	265.148
16	16	:A24.0420	Gỗ ván (m3	0.30959
17	17	:A24.0175	Cát mịn	N m3	73.4145
18	18	:A24.0226	Cọc tre	2 m	2962.47
19	19	:A24.0183	Cây chôn	cây	44.0138


KHỐI LƯỢNG		
455.25		455.25
579.20		579.20
456881		456881.15
367398.85		367398.85
301.52		301.52
246655.39		246655.39
236.83		236.83
316.34		316.34
364.47		364.47
22775.31		22775.31
485.42		485.42
15.58		15.58
11.35		11.35
74.69		74.69
265		265.15
0.31		0.31
73.41		73.41
2962.47		2962.47
44		44.01

3. BÀI TẬP

3.1. Bài tập 1

Thực hiện trước khi làm lại dự án 2. Chú ý là làm theo hướng dẫn của từng bài tập nhỏ với VBA trong Excel


<p>(1) Chương trình con trong VBA</p>	<ul style="list-style-type: none"> ◆ Khởi động VBA IDE trong Excel ◆ Trong dự án mới, tạo mô-đun có tên M1 
---	---





- ◆ Nhập đoạn mã sau trong mô-đun M1:

```
Option Explicit
Dim a As Double
a = 123.45
Debug.Print "a = ",a
```

- ◆ Hãy sửa lỗi để chương trình có thể in ra giá trị của biến a trong cửa sổ Immediate.



	
<p>(2) Ý nghĩa của từ khóa Private và Public</p>	<ul style="list-style-type: none"> ◆ Khởi động VBA IDE trong Excel ◆ Trong dự án mới, tạo 2 mô-đun có tên M1 và M2 ◆ Nhập đoạn mã sau trong mô-đun M1: <pre style="background-color: #f9f9f9; padding: 5px;">Option Explicit Private a As Double a = 123.45</pre> ◆ Nhập đoạn mã sau trong mô-đun M2: <pre style="background-color: #f9f9f9; padding: 5px;">Option Explicit Sub Test1 () Debug.Print a End Sub</pre> ◆ Hãy sửa lỗi để chương trình có thể in ra giá trị của biến a trong cửa sổ Immediate mà không cần khai báo biến a trong mô-đun M2.
<p>(3) Làm quen với kiểu dữ liệu người dùng</p>	<ul style="list-style-type: none"> ◆ Khởi động VBA IDE trong Excel ◆ Trong dự án mới, tạo 2 mô-đun có tên M1 và M2 ◆ Nhập đoạn mã sau trong mô-đun M1: <pre style="background-color: #f9f9f9; padding: 5px;">Option Explicit Private Type DamBTCT ChieuDai As Double ChieuCao As Double End Type</pre> ◆ Nhập đoạn mã sau trong mô-đun M2:

	<pre>Option Explicit Sub Test1() Dim Dam As DamBTCT Dam.ChieuDai = 15 Dam.ChieuCao = 0.85 End Sub</pre> <p>◆ Hãy sửa lỗi và bổ sung để chương trình có thể in ra giá trị của biến Dam trong cửa sổ Immediate từ mô-đun M2.</p>
<p>(4) Làm quen với dữ liệu kiểu mảng</p>	<p>◆ Khởi động VBA IDE trong Excel ◆ Trong dự án mới, tạo mô-đun có tên M1 ◆ Nhập đoạn mã sau trong mô-đun M1(chú ý dấu chấm phẩy “;”):</p> <pre>Option Explicit Sub Test1() Debug.Print "Mang A: "; a(0); a(1); a(2); a(3) Debug.Print "Mang B: "; b(1); b(2); b(3); b(4) End Sub</pre> <p>◆ Không thay đổi nội dung của các lệnh hiện có trong Test1, hãy sửa lỗi và bổ sung để in ra trong cửa sổ Immediate như sau:</p> 
<p>(5) Tạo hàm mới cho Excel</p>	<p>◆ Làm lại ví dụ theo hướng dẫn của mục 3.2.3 trang 113</p>

3.2. Bài tập 2

- Tạo một vùng bảng tính liên tục chứa các số có giá trị ngẫu nhiên từ -100 đến +100 (số hàng và cột là bất kỳ).
- Lập chương trình thực hiện nội dung sau:
 - Yêu cầu người dùng chọn vùng dữ liệu bất kỳ bằng mouse
 - Với vùng dữ liệu được chọn, tìm các số âm (<0) và thực hiện việc thay đổi định dạng cho chúng như sau:
 - Thay đổi màu hiện tại của chữ sang **màu đỏ**.
 - Đổi kiểu chữ hiện tại sang **kiểu đậm** (Bold).

Font		Alignment		Number				
fx		=RANDBETWEEN(-100,100)						
D	E	F	G	H	I	J	K	L
-55	-50	-24	64	93	98	-86	-8	
-82	-18	88	-95	53	-57	-29	-93	
67	-17	-64	30	27	-63	47	78	
-44	-26	94	-89	76	1	62	39	
17	14	-70	29	89	38	-70	59	
-45	-95	-24	-51	-61	66	-47	48	
56	-57	46	-15	6	97	-55	33	
-68	-73	34	-36	72	-45	54	-75	
-21	6	85	-53	67	-9	-36	92	
-99	14	-56	94	-69	-77	8	-64	
4	94	-88	22	-88	92	58	-13	
-63	40	75	40	-85	86	93	-2	
39	-23	25	-89	5	-27	-32	-40	
97	-45	35	-93	88	-86	-9	-64	
72	-34	-11	-49	31	-9	90	-6	
1	-25	-59	21	48	-46	100	-50	
-24	-47	15	-7	-8	-36	-48	-12	
-95	-13	4	-98	-12	-47	25	-39	
39	-63	-17	74	58	-51	86	-54	


- Hướng dẫn cơ bản:
 - Khởi động VBA IDE trong Excel.
 - Tạo ra một module
 - Dùng hàm InputBox của đối tượng Application để làm giao diện nhập dữ liệu
 - Đọc dữ liệu từ bảng tính: mục 7.4.1 trang 156
 - Định dạng Cells (thay đổi màu chữ, kiểu chữ, ...): mục 2.2.2 trang 108

3.3. Bài tập 3

- Xây dựng chương trình là một hàm mới cho phép tra bảng một chiều theo thuật toán nội suy tuyến tính.
- Bảng cần tra là bảng một chiều (như hình dưới), chữ màu đỏ là chỉ số, tương ứng dòng phía dưới là giá trị cần tìm.

Chỉ số	1	3	5	7	9	11
Giá trị	78.99	26.19	59.67	3.81	73.32	31.83

- Tham số đưa vào hàm: giá trị cần tra, vùng dữ liệu của bảng (ví dụ: B2:C7, giá trị ô B2 là 1 và giá trị ô C7 là 31.83). Ví dụ tên hàm là TraBang1 và dữ liệu bảng tra là ô B2:C7 thì sử dụng như sau:



B	C	D	E	F	G
1	3	5	7	9	11
78.99	26.19	59.67	3.81	73.32	31.83

=TraBang1(3,C2:G3)


- Kết quả trả về là số tra được từ vùng dữ liệu trên.
 - Ví dụ cần tra cứu giá trị trong bảng với chỉ số = 3 thì kết quả trả về là 26.19
 - Ví dụ cần tra cứu giá trị trong bảng với chỉ số = 3.5 thì kết quả trả về là 34.56 (nội suy tuyến tính).
- Thông báo bằng MsgBox nếu chỉ số đưa vào không hợp lệ (nằm ngoài phạm vi chỉ số của bảng dữ liệu).
- Hướng dẫn cơ bản:
 - Khởi động VBA IDE trong Excel.
 - Tạo ra một module
 - Cách tạo hàm mới và sử dụng: mục 3.2.3 trang 113
 - Đọc dữ liệu theo hàng, cột: mục 7.4.2 trang 156

DỰ ÁN III: XÂY DỰNG CHƯƠNG TRÌNH NHẬP DỮ LIỆU VÀ TỰ ĐỘNG TẠO BẢN VẼ BỐ TRÍ CHUNG TRỤ CẦU THÂN ĐẶC TRONG AUTOCAD

1. NỘI DUNG DỰ ÁN

1.1. Mô tả yêu cầu

Xây dựng chương trình nhập dữ liệu và tự động tạo bản vẽ bố trí chung trụ cầu thân đặc trong AutoCAD.



1.2. Kịch bản sử dụng và kết quả chương trình

Việc hình dung ra một kịch bản sử dụng từ phía người dùng là rất cần thiết trước khi xây dựng một chương trình. Trong chương trình này, có thể vạch ra một kịch bản như sau:

1. Người dùng chạy chương trình và nhập các dữ liệu trên một cửa sổ.
2. Sau khi nhập dữ liệu, người dùng chọn nút chức năng để yêu cầu chương trình vẽ dựa trên các dữ liệu đã nhập.
3. Chương trình AutoCAD yêu cầu người dùng chọn một điểm trên màn hình để bắt đầu vẽ.
4. Cuối cùng, bản vẽ bố trí chung trụ được tạo ra theo yêu cầu.

1.3. Cấu trúc dự án

1.3.1. Bản vẽ mẫu xuất phát

Việc tạo bản vẽ mẫu xuất phát nhằm tạo ra các định dạng cần thiết (các lớp, kiểu đường, kiểu chữ,...)

Bản vẽ mẫu tương ứng với dự án này bao gồm các lớp sau:

- Lớp “Netdam”: phục vụ vẽ các đường bao kết cấu.
- Lớp “Netdut”: phục vụ vẽ các đường nét khuất (nếu có).

- Lớp “Kichthuoc”: phục vụ vẽ các đường kích thước trong bản vẽ.

★ *Tham khảo bản vẽ mẫu: “StartingDrawing.dwg”.*

1.3.2. Giao diện người sử dụng (UserForm)

Giao diện người sử dụng hay UserForm được thiết kế để nhập dữ liệu cho các kích thước của trụ. Các thành phần trên giao diện cần có tính minh họa tốt để định hướng người dùng. Ngoài ra, nó còn là nơi để người dùng quyết định khi nào thì tiến hành vẽ (xử lý sự kiện khi người dùng ra lệnh).

1.3.3. Khối chương trình (Module)

Các mã lệnh phục vụ quá trình vẽ được chứa trong một Module. Có thể chia các mã lệnh thành những phần sau:

- Mã lệnh khai báo kiểu dữ liệu mô tả trụ.
- Các mã khai báo dữ liệu tổng thể.
- Các mã lệnh thực hiện vẽ được chia thành các chức năng nhỏ (các chương trình con):
 - Vẽ hình chiếu đứng.
 - Vẽ hình chiếu cạnh.
 - Vẽ hình chiếu bằng.
 - Vẽ toàn thể trụ (chương trình con gọi các chức năng vẽ hình chiếu).

1.4. Nội dung kiến thức VBA liên quan đến dự án

- Kiểu dữ liệu mảng và khai báo mảng.

✓ Tham khảo: Giáo trình, chương III, mục 5.4, trang 28→29.

- Khai báo kiểu dữ liệu tự định nghĩa.

✓ Tham khảo: Giáo trình, chương III, mục 6.3, trang 37.

- Tổ chức chương trình theo hệ thống các mô đun chuẩn (Module).

✓ Tham khảo: Giáo trình, chương III, mục 10, trang 58→59.

- Làm việc với chương trình con.

✓ Tham khảo: Giáo trình, chương III, mục 9, trang 50→58.

- Tạo giao diện người sử dụng (UserForm).

✓ Tham khảo: Giáo trình, chương III, mục 10, trang 59→75.

- Nhập dữ liệu từ dòng lệnh của AutoCAD – Đối tượng Utility.

✓ Tham khảo: Giáo trình, chương V, mục 5.1.5, trang 212.

- Làm việc với lớp trong AutoCAD.

✓ Tham khảo: Giáo trình, chương V, mục 5.5, trang 260→264.

- Tạo đối tượng dạng đường thẳng trong AutoCAD.

✓ Tham khảo: Giáo trình, chương V, mục 5.2.4, trang 226→229.

- Tạo đối tượng dạng đường cong trong AutoCAD.

✓ Tham khảo: Giáo trình, chương V, mục 5.25, trang 229→232.

- Hiệu chỉnh đối tượng hình học trong AutoCAD.

✓ Tham khảo: Giáo trình, chương V, mục 5.2, trang 244→260.

- Thao tác với đường kích thước trong AutoCAD.

✓ Tham khảo: Giáo trình, chương V, mục 5.7, trang 267→275.

1.5. Những lưu ý khác

Các hướng dẫn của dự án này được thực hiện trên phiên bản AutoCAD 2007 trở lên.

2. TRÌNH TỰ THỰC HIỆN DỰ ÁN

2.1. Tạo và lưu bản vẽ mẫu xuất phát

2.2. Tạo và lưu dự án VBA

Nhấn tổ hợp phím **Alt+F11** để mở môi trường phát triển tích hợp VBA.

Dự án nên được lưu dưới dạng độc lập (*.dvb).

2.3. Xây dựng giao diện người sử dụng

Tạo một UserForm với tên (Name) là **Cuasochinh**.

The screenshot shows a VBA UserForm titled "Nhập kích thước và vẽ bố trí chung trụ than đặc". On the left, there is a technical drawing of a boiler structure with dimensions labeled: GD (top diameter), GC (middle diameter), GB (bottom diameter), b1 (width of the top section), h1 (height of the top section), b2 (width of the middle section), h2 (height of the middle section), a1 (width of the bottom section), and a2 (width of the bottom section). A red box labeled "KcDim" highlights a dimension line. On the right, there is a list of input fields for these dimensions:

Kích thước than trụ dọc cao a1 (a1 = 2R)	2.2
Kích thước than trụ ngang cao b1	5.3
Chiều cao than trụ h1	7
Kích thước be trụ dọc cao a2	4.2
Kích thước be trụ ngang cao b2	8.2
Chiều cao be trụ h2	2
Khoảng cách giữa hai hình chiều GD-GB	12.5
Khoảng cách giữa hai hình chiều GD-GC	8.2
Khoảng cách ghi kích thước (KcDim)	0.7

At the bottom right, there is a label "Đơn vị nhập: m" and a button "Vẽ trụ cao".

Một số thành phần chính cần chú ý:

Tên (Name)	Loại điều khiển	Mô tả
Image1	Image	Chứa hình minh họa biểu diễn dữ liệu nhập (trong thuộc tính Picture)
txta1	Textbox	Dữ liệu kích thước thân trụ dọc cầu a1 (mặc định 2.2)
txtb1	Textbox	Dữ liệu kích thước thân trụ ngang cầu b1 (mặc định 5.3)
txth1	Textbox	Dữ liệu chiều cao thân trụ h1 (mặc định 7)
txta2	Textbox	Dữ liệu kích thước bệ trụ dọc cầu a2 (mặc định 4.2)
txtb2	Textbox	Dữ liệu kích thước bệ trụ ngang cầu b2 (mặc định 8.2)
txth2	Textbox	Dữ liệu chiều cao trụ h2 (mặc định 2)
txtKcD_B	Textbox	Dữ liệu khoảng cách hình chiếu đứng - bằng (mặc định 12.5)
txtKcD_C	Textbox	Dữ liệu khoảng cách hình chiếu đứng - cạnh (mặc định 8.2)
txtKcDi_m	Textbox	Dữ liệu khoảng cách ghi kích thước (mặc định 0.7)
btnVe	CommandButton	Tiếp nhận lệnh yêu cầu vẽ từ phía người dùng Tiêu đề (caption): " Ve tru cau "

Ngoài ra, trên giao diện người dùng còn có các nhãn (Label) nhằm giải thích, mô tả dữ liệu nhập.

★ **Kiến thức tham khảo**

✓ Tạo giao diện người sử dụng (UserForm): Giáo trình, chương III, mục 10, trang 59→75.

2.4. Xây dựng khối chương trình dạng khung

Thêm một mô-đun chuẩn (Module) vào dự án với tên được đặt là **Chuongtrinh**.

2.4.1. Viết mã lệnh khai báo kiểu dữ liệu mô tả trụ

```
'KHAIBAO KIEUDU LIEU MO TA TRU
Public Type T_Tru
    a1 As Double 'kich thuc than tru doc cau
    b1 As Double 'kich thuc than tru ngang cau
    h1 As Double 'chieu cao than tru
    a2 As Double 'kich thuc be tru doc cau
    b2 As Double 'kich thuc be tru ngang cau
    h2 As Double 'chieu cao be tru
End Type
```

★ **Kiến thức tham khảo**

✓ Khai báo kiểu tự định nghĩa: Giáo trình, chương III, mục 5.8, trang 31.

2.4.2. Khai báo các dữ liệu toàn cục liên quan đến bố trí các bản vẽ hình chiếu

```
'KHAIBAO CAC DU LIEU TOAN CUC
Public Const Pi = 3.14159265358979 'hang so Pi
Public kcDim As Double 'khoang cach tu diem goc -> kich thước
Public kcD_C As Double 'khoang cach hcDung -> hcCanh
Public kcD_B As Double 'khoang cach hcDung -> hcBang
```

Ý nghĩa của các dữ liệu khai báo như sau:

- Khai báo hằng số **Pi** nhằm phục vụ cho các phép tính đổi góc từ độ (°) sang radian (rad) hoặc ngược lại.
- Biến **kcDim** mô tả khoảng cách từ điểm gốc kích thước tới nơi hiển thị chữ số kích thước.
- Biến **kcD_C** mô tả khoảng cách từ điểm gốc vẽ hình chiếu đứng đến gốc vẽ hình chiếu cạnh.
- Biến **kcD_B** mô tả khoảng cách từ điểm gốc vẽ hình chiếu đứng đến gốc vẽ hình chiếu bằng.

2.4.3. Khai báo các mẫu chương trình con trong mô đun đã tạo

- Chương trình con vẽ hình chiếu đứng (**VeHCDung**)

Chương trình này có 2 tham số: tham số **Tru** chứa dữ liệu về trụ đang xét, tham số **Gocve** là dữ liệu điểm vẽ hình chiếu đứng. Dữ liệu điểm vẽ trong AutoCAD thực chất là một mảng một chiều 3 phần tử kiểu Double; tuy nhiên, trong trường hợp này **Gocve** có kiểu **Variant** để tiện cho việc gán tham số cho chương trình con (trong chương trình con không chấp nhận tham số là một mảng có kích thước cố định). Khi khai báo như vậy, tham số **Gocve** cần được ngầm hiểu là một mảng 3 phần tử kiểu **Double**.

```
'Chương trình con ve hình chiếu đứng
Private Sub VeHCDung(Tru As T_Tru, Gocve As Variant)
End Sub
```

★ Kiến thức tham khảo

- ✓ Dữ liệu kiểu Variant: Giáo trình, chương III, mục 5.7, trang 30-31.
- ✓ Chương trình con dạng thủ tục (Sub): Giáo trình, chương III, mục 9.2, trang 51.

- Chương trình con vẽ hình chiếu cạnh (**VeHCCanh**)

Chương trình này có 2 tham số: tham số **Tru** chứa dữ liệu về trụ đang xét, tham số **Gocve** là dữ liệu điểm vẽ hình chiếu cạnh.

```
'Chương trình con ve hình chiếu cạnh
Private Sub VeHCCanh(Tru As T_Tru, Gocve As Variant)
End Sub
```

- Chương trình con vẽ hình chiếu bằng (**VeHCBang**)

Chương trình này có 2 tham số: tham số **Tru** chứa dữ liệu về trụ đang xét, tham số **Gocve** là dữ liệu điểm vẽ hình chiếu bằng.

```
'Chương trình con ve hình chiếu bằng
Private Sub VeHCBang(Tru As T_Tru, Gocve As Variant)
End Sub
```

- Chương trình con vẽ toàn bộ trụ (**VeTru**)

Chương trình này thực chất có nhiệm vụ triệu gọi các chương trình con vẽ các hình chiếu và gán các điểm vẽ tương ứng để tạo nên bản vẽ bố trí chung bao gồm các bản vẽ hình chiếu.

Chương trình có 2 tham số: tham số **Tru** chứa dữ liệu về trụ đang xét, tham số **Diemve** là dữ liệu điểm vẽ trụ (điểm này sẽ lấy trùng với điểm vẽ hình chiếu đứng).

```
'Chương trình con ve trụ - tat ca cac hình chiếu
Public Sub VeTru(Tru As T_Trụ, Diemve As Variant)
End Sub
```

2.5. Viết mã lệnh chi tiết

2.5.1. Mã lệnh xử lý sự kiện trên giao diện người dùng

Trên giao diện người dùng, sự kiện cần quan tâm là khi người dùng kích chuột vào nút lệnh **btnVe** (có tiêu đề “**Ve trụ cau**”). Vì vậy, các mã lệnh xử lý tương ứng với thủ tục sự kiện **Click** của nút lệnh **btnVe**. Các lệnh xử lý bao gồm:

- Khai báo các biến nhằm lưu trữ dữ liệu mà người dùng nhập từ **UserForm**.
- Đưa các dữ liệu người dùng đã nhập từ **UserForm** vào các biến trong chương trình xử lý.
- Yêu cầu người dùng chọn điểm bắt đầu vẽ trên màn hình.
- Gọi chương trình vẽ tương ứng với các dữ liệu đã nhập (gọi thủ tục **Vetru** đã được khai báo ở trên).

Mở **UserForm “Cuasochinh”**; chọn đúp chuột vào nút lệnh **btnVe** để viết mã lệnh cho thủ tục sự kiện **btnVe_Click**. Mã lệnh được trình bày như dưới đây:

```
1 Private Sub btnVe_Click()
2   ' Thiet lap khoang cach trong trình bày bản vẽ
3     Chuongtrinh.kcD_B = txtKcD_B.Value
4     Chuongtrinh.kcD_C = txtKcD_C.Value
5     Chuongtrinh.kcDim = txtKcDim.Value
6   ' Nhan du lieu kích thước trụ từ giao diện người dùng
7     Dim Tru As Chuongtrinh.T_Trụ
8     Tru.a1 = txta1.Value: Tru.b1 = txtb1.Value: Tru.h1 = txth1.Value
9     Tru.a2 = txta2.Value: Tru.b2 = txtb2.Value: Tru.h2 = txth2.Value
10  ' An cua so nhập dữ liệu và chọn điểm vẽ hình chiếu dùng
11    Me.Hide
12    Dim Diemchon As Variant
13    ThisDrawing.Utility.Prompt "Chọn điểm bắt đầu vẽ: "
14    Diemchon = ThisDrawing.Utility.GetPoint()
15  ' Goi lenh vẽ
16    Chuongtrinh.VeTru Tru, Diemchon
17 End Sub
```


★ Kiến thức tham khảo

- ✓ Tạo sự kiện cho các điều khiển: Giáo trình, chương III, mục 11.1.4, trang 65.
- ✓ Thuộc tính của điều khiển Textbox (dòng lệnh 3→9): Giáo trình, chương III, mục 11.3, trang 69.
- ✓ Thuộc tính Value của điều khiển Textbox trả về giá trị kiểu Variant tương ứng với dữ liệu chuỗi đã nhập tương ứng ở điều khiển đó.
- ✓ Phương thức Hide của UserForm (dòng lệnh 11): Giáo trình, chương III, mục 11.2, trang 67.
- ✓ Nhập dữ liệu từ dòng lệnh của AutoCAD (dòng lệnh 13→14): Giáo trình, chương V, mục 5.1.5, trang 213.

✓ Cách thức gọi chương trình con (dòng lệnh 16): Giáo trình, chương III, mục 9.5, trang 57.

2.5.2. Chương trình con vẽ toàn bộ trụ

Mở Module “**Chuongtrinh**” và viết mã lệnh chi tiết sau vào chương trình con **VeTru** đã tạo:



```
'Chương trình con ve tru - tat ca cac hinh chieu
Public Sub VeTru(Tru As T Tru, Diemve As Variant)
'Diem ve cua tru tuong ung voi goc ve cua HC dung
Dim DiemHchieu As Variant
' Ve hinh chieu dung
DiemHchieu = Diemve
VeHCDung Tru, DiemHchieu
' Ve hinh chieu bang
DiemHchieu = Diemve: DiemHchieu(1) = Diemve(1) - kcD_B
VeHCBang Tru, DiemHchieu
' Ve hinh chieu canh
DiemHchieu = Diemve: DiemHchieu(0) = Diemve(0) + kcD_C
VeHCCanh Tru, DiemHchieu
End Sub
```


2.5.3. Chương trình vẽ hình chiếu đứng

Tìm tới chương trình con **VeHCDung** đã được tạo trong mô đun **Chuongtrinh**.

Viết thêm các mã lệnh vào trong thân chương trình. Các mã lệnh này bao gồm:

2.5.4. Mã lệnh vẽ các đường bao của hình chiếu đứng.

Minh họa trình tự vẽ



```
1 Dim L(1 To 5) As AcadLine
2 Dim SP As Variant, EP As Variant
3 'VE CAC DUONG THANG
4 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Netdam")
5 'line1
6 SP = Gocve: SP(0) = SP(0) - Tru.b1 / 2
7 EP = SP: EP(0) = EP(0) + Tru.b1
8 Set L(1) = ThisDrawing.ModelSpace.AddLine(SP, EP)
9 'line 2
10 EP = SP: EP(1) = EP(1) - Tru.h1
11 Set L(2) = ThisDrawing.ModelSpace.AddLine(SP, EP)
12 'line 3
13 SP = EP: SP(0) = Gocve(0) - Tru.b2 / 2
14 EP(0) = SP(0) + Tru.b2
15 Set L(3) = ThisDrawing.ModelSpace.AddLine(SP, EP)
16 'line 4
17 EP = SP: EP(1) = EP(1) - Tru.h2
18 Set L(4) = ThisDrawing.ModelSpace.AddLine(SP, EP)
19 'line 5
20 SP = EP: EP(0) = EP(0) + Tru.b2
21 Set L(5) = ThisDrawing.ModelSpace.AddLine(SP, EP)
22 'lay doi xung tu L - 2,4 qua truc y
23 SP = Gocve: EP = Gocve: EP(1) = EP(1) + 100
24 L(2).Mirror SP, EP: L(4).Mirror SP, EP
```

★ Kiến thức tham khảo

- ✓ Thiết lập một lớp (layer) thành lớp hiện thời (dòng lệnh 4): Giáo trình, chương V, mục 5.5.3, trang 262.
- ✓ Vẽ một đoạn thẳng trong AutoCAD (dòng lệnh 8,11,15,18,21): Giáo trình, chương V, mục 5.2.4, trang 227.
- ✓ Lấy đối xứng một đối tượng trong AutoCAD (dòng lệnh 22→24): Giáo trình, chương V, mục 5.4.1, trang 248.

★ **Chú ý:**

Trong đoạn mã lệnh trên, dữ liệu tương ứng với điểm bắt đầu (SP) và điểm kết thúc của một đoạn thẳng được khai báo với kiểu **Variant** nhưng cần được ngầm hiểu là một mảng một chiều 3 phần tử kiểu **Double**. Việc khai báo các dữ liệu trên với kiểu **Variant** cho phép thực hiện phép gán dễ dàng, điều mà nếu khai báo theo kiểu thông thường không thể thực hiện được.

Ví dụ:

Nếu khai báo: **Dim SP(0 to 2) as Double, EP(0 to 2) as Double**

Thì phép gán **EP=SP** sẽ không thể thực hiện được.

★ **Sau khi viết mã lệnh vẽ đường bao, có thể chạy thử kết quả của đoạn mã đó bằng cách: chọn vào UserForm Cuasochinh; nhấn F5; nhấn tiếp vào nút lệnh "Ve tru cau" trên giao diện nhập dữ liệu và quan sát kết quả trên AutoCAD.**

2.5.5. Mã lệnh ghi kích thước trên hình chiếu đứng (viết tiếp theo các mã lệnh vẽ đường bao)

```

25 'GHI KICH THUOC
26 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Kichthuoc")
27 Dim Vitri As Variant 'diem dat kich thuoc
28 'kt1
29 SP = L(2).EndPoint: EP = SP: EP(0) = SP(0) + Tru.b1
30 Vitri = SP: Vitri(1) = SP(1) + kcDim
31 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, 0
32 'kt2
33 SP = L(5).StartPoint: EP = L(5).EndPoint
34 Vitri = SP: Vitri(1) = SP(1) - kcDim
35 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, 0
36 'kt3
37 SP = L(1).EndPoint: EP = L(3).EndPoint
38 Vitri = EP: Vitri(0) = Vitri(0) + kcDim
39 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, Pi / 2
40 'kt4
41 SP = L(3).EndPoint: EP = L(5).EndPoint
42 Vitri = EP: Vitri(0) = Vitri(0) + kcDim
43 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, Pi / 2

```

★ **Kiến thức tham khảo**

- ✓ Tạo một đường kích thước DimRotated trong AutoCAD (dòng lệnh 43): Giáo trình, chương V, mục 5.7.2, trang 270.
- ✓ Thuộc tính StartPoint (hoặc EndPoint) của một đoạn thẳng trả về một dữ liệu Variant tương ứng với tọa độ của điểm đó (mảng một chiều 3 phần tử kiểu Double).

★ **Sau khi viết đoạn mã lệnh trên vào chương trình con VeHCDung, có thể chạy thử kết quả của đoạn mã đó bằng cách: chọn vào UserForm Cuasochinh; nhấn F5; nhấn tiếp vào nút lệnh "Ve tru cau" trên giao diện nhập dữ liệu và quan sát kết quả trên AutoCAD.**

2.5.6. Chương trình vẽ hình chiếu cạnh

Tìm tới chương trình con **VeHCCanh** đã được tạo trong mô đun **Chuongtrinh**. Các mã lệnh thêm vào trong thân chương trình bao gồm:


2.5.7. Mã lệnh vẽ các đường bao hình chiếu cạnh.

Viết thêm các mã lệnh tương tự như với **VeHCDung**. Các mã lệnh này có thể soạn thảo nhanh bằng cách **Copy** mã lệnh tương ứng trong chương trình **VeHCDung** rồi thực hiện thao tác **tìm và thay thế** (**Find and Replaced** – hoặc tổ hợp phím **Ctrl + H**). Dữ liệu tương ứng với các kích thước ngang cầu sẽ được thay bằng các dữ liệu tương ứng với kích thước dọc cầu (**Tru.b1** thay bằng **Tru.a1**, **Tru.b2** thay bằng **Tru.a2**).

```

1 Dim L(1 To 5) As AcadLine
2 Dim SP As Variant, EP As Variant
3 'VE CAC DUONG THANG
4 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Netdam")
5 'line1
6 SP = Gocve: SP(0) = SP(0) - Tru.a1 / 2
7 EP = SP: EP(0) = EP(0) + Tru.a1
8 Set L(1) = ThisDrawing.ModelSpace.AddLine(SP, EP)
9 'line 2
10 EP = SP: EP(1) = EP(1) - Tru.h1
11 Set L(2) = ThisDrawing.ModelSpace.AddLine(SP, EP)
12 'line 3
13 SP = EP: SP(0) = Gocve(0) - Tru.a2 / 2
14 EP(0) = SP(0) + Tru.a2
15 Set L(3) = ThisDrawing.ModelSpace.AddLine(SP, EP)
16 'line 4
17 EP = SP: EP(1) = EP(1) - Tru.h2
18 Set L(4) = ThisDrawing.ModelSpace.AddLine(SP, EP)
19 'line 5
20 SP = EP: EP(0) = EP(0) + Tru.a2
21 Set L(5) = ThisDrawing.ModelSpace.AddLine(SP, EP)
22 'lay doi xung tu L - 2,4 qua truc y
23 SP = Gocve: EP = Gocve: EP(1) = EP(1) + 100
24 L(2).Mirror SP, EP: L(4).Mirror SP, EP

```




2.5.8. Mã lệnh ghi kích thước cho hình chiếu cạnh.

```

25 'GHI KICH THUOC
26 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Kichthuoc")
27 Dim Vitri As Variant 'diem dat kich thuoc
28 'kt1
29 SP = L(2).EndPoint: EP = SP: EP(0) = SP(0) + Tru.a1
30 Vitri = SP: Vitri(1) = SP(1) + kcDim
31 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, 0
32 'kt2
33 SP = L(5).StartPoint: EP = L(5).EndPoint
34 Vitri = SP: Vitri(1) = SP(1) - kcDim
35 ThisDrawing.ModelSpace.AddDimRotated SP, EP, Vitri, 0

```



★ Sau khi viết các đoạn mã tương ứng với vẽ hình chiếu cạnh, việc kiểm tra thử đoạn mã lệnh vừa tạo nên được thực hiện. Cách thức kiểm tra giống như đã thực hiện với các đoạn mã vẽ hình chiếu đứng.

2.5.9. Chương trình vẽ hình chiếu bằng


Tìm tới chương trình con **VeHCBang** đã được tạo trong Module **Chuongtrinh**. Các mã lệnh thêm vào trong thân chương trình bao gồm:

2.5.10. Mã lệnh vẽ các đường bao trên của hình chiếu bằng.

```

1 Dim L(1 To 3) As AcadLine
2 Dim SP As Variant, EP As Variant
3 'VE CAC DUONG BAO
4 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Netdam")
5 'line1
6 SP = Gocve: SP(0) = Gocve(0) - Tru.b2 / 2: SP(1) = Gocve(1) + Tru.a2 / 2
7 EP = SP: EP(0) = SP(0) + Tru.b2
8 Set L(1) = ThisDrawing.ModelSpace.AddLine(SP, EP)
9 'line2
10 EP = SP: EP(1) = SP(1) - Tru.a2
11 Set L(2) = ThisDrawing.ModelSpace.AddLine(SP, EP)
12 'line3
13 SP(0) = Gocve(0) - Tru.b1 / 2 + Tru.a1 / 2: SP(1) = Gocve(1) + Tru.a1 / 2
14 EP = SP: EP(0) = SP(0) + Tru.b1 - Tru.a1
15 Set L(3) = ThisDrawing.ModelSpace.AddLine(SP, EP)
16 'Arc4
17 SP(1) = SP(1) - Tru.a1 / 2
18 Dim A4 As AcadArc
19 Set A4 = ThisDrawing.ModelSpace.AddArc(SP, Tru.a1 / 2, Pi / 2, 3 * Pi / 2)
20 'Lay doi xung L1,3 qua truc x
21 SP = Gocve: EP = Gocve: EP(0) = EP(0) + 100
22 L(1).Mirror SP, EP: L(3).Mirror SP, EP
23 'Lay doi xung L2,A4 qua truc y
24 EP = SP: EP(1) = EP(1) + 100
25 L(2).Mirror SP, EP: A4.Mirror SP, EP

```



★ Kiến thức tham khảo

✓ Tạo một đối tượng cung tròn (Arc) trong AutoCAD: Giáo trình, chương V, mục 5.2.2, trang 230-231.

★ Sau khi thêm đoạn mã lệnh trên vào chương trình **VeHCBang**, việc chạy thử nên được tiếp tục thực hiện như các bước ở trên.

2.5.11. Mã lệnh ghi kích thước.

```

26 'GHI KICH THUOC
27 ThisDrawing.ActiveLayer = ThisDrawing.Layers("Kichthuoc")
28 Dim Vitri As Variant 'diem dat kich thuoc
29 'kt1 - cung tron
30 SP = A4.Center: SP(0) = SP(0) - A4.Radius
31 ThisDrawing.ModelSpace.AddDimRadial A4.Center, SP, -(A4.Radius)

```

★ Kiến thức tham khảo

✓ Tạo đường kích thước bán kính DimRadial trong AutoCAD (dòng lệnh 31): Giáo trình, chương V, mục 5.7.2, trang 273.

★ *Tiếp tục chạy thử cho mã lệnh vừa tạo.*

2.6. Chạy thử chương trình và kiểm tra kết quả


3. BÀI TẬP

3.1. Bài tập thực hành

3.1.1. Bài tập 1

3.1.2. Mô tả

Tạo bản vẽ bố trí móng cọc trên mặt bằng như hình vẽ dưới đây.




3.1.3. Hướng dẫn

- Tạo giao diện người sử dụng để nhập các dữ liệu cần thiết.
- Tạo một mô đun để chứa chương trình vẽ.
- Chương trình vẽ sẽ bao gồm các loại mã lệnh sau:
 - Mã lệnh vẽ các đường bao
 - Vẽ hình đường bao bộ cọc.
 - Vẽ đường bao cọc.
 - Mã lệnh ghi kích thước.
 - Các kích thước thẳng.
 - Ghi kích thước bán kính cọc hoặc đường kính cọc.

3.1.4. Bài tập 2

3.1.5. Mô tả

Tạo bản vẽ mặt cắt dầm như hình vẽ dưới đây.



3.1.6. Hướng dẫn

- Tạo giao diện người sử dụng để nhập các dữ liệu cần thiết.
- Tạo một mô đun để chứa chương trình vẽ.
- Chương trình vẽ sẽ bao gồm các loại mã lệnh sau:
 - Mã lệnh vẽ các đường bao
 - Mã lệnh ghi kích thước.

3.2. Câu hỏi ôn tập

DỰ ÁN IV: CHƯƠNG TRÌNH VẼ TRẮC ĐỌC TỰ NHIÊN TRÊN AUTOCAD VỚI SỐ LIỆU ĐỌC TỪ FILE EXCEL


1. NỘI DUNG DỰ ÁN

1.1. Mô tả yêu cầu

- Tên dự án: Chương trình vẽ trắc dọc tự nhiên trên AutoCAD với số liệu đọc từ file Excel.
- Số liệu của trắc dọc tự nhiên gồm: Mức so sánh, Tên cọc, Cao độ, Khoảng cách lẻ được lưu trữ trong file Excel theo định dạng bên dưới:

	A	B	C	D
1	MSS	2		
2	Tên cọc	Cao độ	Khoảng cách lẻ	
3	KM0	12.5	0	
4	1	15	3	
5	2	14.3	5	
6	3	13.5	4	
7	4	12.8	5	
8	TD1	13.4	5	
9	P1	14	6	
10	TC1	15	5	
11	8	15.6	4	
12	9	14.3	5	
13	10	12	3	
14	11	14	2	
15	12	11	3	
16				

- Xây dựng phần mềm tự động đọc số liệu từ file Excel để vẽ trắc dọc tự nhiên trên AutoCAD. Người dùng chọn vị trí muốn vẽ và sẽ có kết quả vẽ như sau:




1.2. Cấu trúc dự án

1.2.1. Dữ liệu ban đầu


- File “td.xlsx” để tại thư mục gốc ổ đĩa C:, chứa dữ liệu trắc dọc ở Sheet1, các Sheet khác không có dữ liệu.

1.2.2. Giao diện người dùng

- Chương trình được sử dụng dưới dạng Macro (xem mục 2.3 trang 108).




- Người dùng cần chọn vị trí cần vẽ trắc ngang, chương trình sẽ tự động vẽ



1.2.3. Khởi chương trình

- Chương trình được đặt trong 01 Module.



1.3. Nội dung kiến thức VBA liên quan đến dự án

- Yêu cầu khai báo biến trước khi sử dụng: mục 2 trang 23.
- Khởi động Excel từ chương trình khác: mục 7.1.4 trang 152
- Khai báo biến cơ bản: mục 5 trang 26 và mục 6 trang 33.
- Khai báo và sử dụng mảng động: mục 6.5 trang 38.
- Khai báo biến có kiểu đối tượng của Excel: mục 7.1.4 trong 152
- Các đối tượng trong AutoCAD, mục 4.1 trang 199
- Cấu trúc If ... Then ... Else ... End If : mục 8.1 trang 42.
- Cấu trúc For ... To ...Next : mục 8.3.1 trang 46.
- Cấu trúc Do Until...Loop : mục 8.4 trang 50.
- Đọc dữ liệu với Cells(): mục 5.2.5 trang 135
- Lấy tọa độ của điểm bất kỳ trên bản vẽ: mục 5.1.5 trang 217
- Khai báo và sử dụng văn bản (Text): mục 5.2.6 trang 232.
- Khai báo và sử dụng lớp (Layer): mục 5.5.1 trang 261
- Khai báo và sử dụng điểm (Point): mục 5.2.3 trang 225.
- Khai báo và vẽ đoạn thẳng: mục 5.2.4 trang 226.
- Khai báo và vẽ đường đa tuyến rút gọn (LightWeightPolyline): mục 5.2.4 trang 228.


1.4. Những lưu ý khác

- Phần hướng dẫn này phù hợp với phiên bản AutoCAD và Excel từ 2007 trở lên.
- Tài liệu tham khảo (mục, trang) được lấy theo Giáo trình môn Tự động hóa Thiết kế cầu đường (bản PDF).
- Có 2 cách để lưu trữ dự án: dạng độc lập ở định dạng DVB hoặc dạng nhúng nằm ngay trong file bản vẽ DWG (tham khảo mục 2.1 trang 190).

2. TRÌNH TỰ THỰC HIỆN DỰ ÁN

2.1. Chạy VBA IDE và tạo Module

- Chạy phần mềm AutoCAD
- Gọi VBA IDE: bấm Alt + F11
- Tạo Module: Chọn Insert → Module




2.2. Lưu trữ dự án dạng DVB

- Chọn menu: File → Save Global1 → Đặt tên: *Ve trac doc tu nhien.dvb*

2.3. Khai báo tham chiếu hệ thống đối tượng của Excel trong AutoCAD

- Chọn menu: Tools → References → Đánh dấu vào mục Microsoft Excel Object Library (Phiên bản 14.0 ứng với Excel 2010)



2.4. Xây dựng giao diện người sử dụng

- Chương trình này không có giao diện dạng UserForm mà thực hiện thông qua cách gọi Macro của AutoCAD (xem mục 2.3 trang 108).

2.5. Xây dựng các chương trình con

2.5.1. Khai báo các biến

Khai báo biến kiểu đối tượng Excel, Workbook và Worksheet.

Kiểm thức tham khảo

- ✓ Các đối tượng cơ bản trong Excel: mục 5.2 trang 123

```

1 Option Explicit
2 'Khai báo biến
3 Dim AppEX As Excel.Application      'Biến ứng dụng Excel
4 Dim WB As Excel.Workbook           'Workbook
5 Dim WS As Excel.Worksheet          'Worksheet

```

Khai báo biến kiểu dữ liệu thông thường

Kiểm thức tham khảo

- ✓ Khai báo biến trong VB: mục 6.2 trang 37
- ✓ Khai báo mảng động: mục 6.5 trang 38

```

7 Dim TongSoCoc As Integer           'Tổng số cọc cần vẽ
8 Dim i As Integer                   'Biến chạy
9 Dim MucSoSanh As Double            'Mức so sánh
10 Dim KCCD As Double                 'Khoảng cách cộng dồn
11 Dim DiemBatDauVe As Variant        'Điểm bắt đầu vẽ
12 Dim SP As Variant                  'Điểm đầu
13 Dim EP As Variant                  'Điểm cuối
14 Dim ToaDoTDTN() As Double          'Mảng Lưu trữ tọa độ trắc dọc
15 Dim KhoangCachLe As String         'Khoảng cách lẻ

```

Khai báo biến kiểu đối tượng AutoCAD

Kiểm thức tham khảo

- ✓ Các đối tượng trong AutoCAD, mục 4.1 trang 199
- ✓ Tạo mới lớp bản vẽ: mục 5.5.1 trang 261

```

17 Dim GhiChu As AcadText             'Ghi chú
18 Dim LopBanVe As AcadLayer          'Lớp bản vẽ

```

2.5.2. Tạo chương trình dạng Macro trong Module

Kiểm thức tham khảo

- ✓ Khi không có từ khóa nào trước Sub, mặc định sẽ là Public: mục 9 trang 50 (tham khảo thêm mục 6 trang 33).
- ✓ Chương trình con dạng thủ tục (Sub): mục 9.2, trang 51.
- ✓ Thu phóng bản vẽ: mục 5.1.4 trang 211

Chương trình con này sẽ thực hiện tất cả các nội dung của dự án và là dạng Macro của AutoCAD.

```

20 Sub VeTracDocTuNhien ()
21 'Gọi lần lượt các chương trình con
22     TaoLopBanVe           'Khởi tạo các lớp bản vẽ
23     DocDuLieuTuExcel     'Đọc dữ liệu từ Excel
24     DienMucSoSanh       'Điền mức so sánh
25     VePolylineMatDat    'Vẽ đường mặt đất tự nhiên
26     VeDuongDongDung     'Vẽ đường đóng đưng
27     DienThongSoTracDoc  'Điền thông số trắc dọc
28     HoanThienTracDoc    'Hoàn thiện trắc dọc
29     MsgBox "Da ve trac doc tu nhien!"
30     ZoomExtents
31 End Sub
    
```

2.5.3. Tạo lớp bản vẽ

Kiến thức tham khảo

- ✓ Tạo mới lớp bản vẽ: mục 5.5.1 trang 261

```

33 Private Sub TaoLopBanVe () 'Tạo lớp bản vẽ và thiết lập màu sắc
34     Set LopBanVe = ThisDrawing.Layers.Add("PLINETDTN") : LopBanVe.Color = acGreen
35     Set LopBanVe = ThisDrawing.Layers.Add("LINEDONGTD") : LopBanVe.Color = acWhite
36     Set LopBanVe = ThisDrawing.Layers.Add("ENTTDTUNHIEN") : LopBanVe.Color = acGreen
37     Set LopBanVe = ThisDrawing.Layers.Add("TEXTTENCOC") : LopBanVe.Color = acBlue
38     Set LopBanVe = ThisDrawing.Layers.Add("TEXTHEADERTN") : LopBanVe.Color = acCyan
39 End Sub
    
```

- Kết quả

S...	Name	O...	Fre...	L...	Color	Linetype
0					wh...	Continu...
	ENTTDTUNHIEN				gr...	Continu...
	LINEDONGTD				wh...	Continu...
	PLINETDTN				gr...	Continu...
	TEXTHEADERTN				cyan	Continu...
	TEXTTENCOC				blue	Continu...

2.5.4. Đọc dữ liệu từ Excel trong AutoCAD

Kiến thức tham khảo

- ✓ Khởi động Excel từ chương trình khác: mục 7.1.4 trang 152
- ✓ Đọc dữ liệu từ Cells(): mục 5.2.5 trang 135
- ✓ Sử dụng Do Until...Loop : mục 8.4 trang 50.

Khởi tạo ứng dụng Excel chạy ở chế độ ẩn.

Gán Workbook chứa dữ liệu trắc dọc cho biến WB và gán Worksheet số 1 cho biến WS.

```

41 Private Sub DocDuLieuTuExcel ()
42     'Khởi tạo ứng dụng Excel
43     Set AppEX = New Excel.Application           'Khởi tạo ứng dụng
44     AppEX.Visible = False                     'Thuộc tính hiển thị của AppExcel (Không hiển thị)
45
46     Set WB = AppEX.Workbooks.Open("c:\td")    'Đường dẫn file dữ liệu Excel
47     Set WS = WB.Worksheets(1)                'Gán WS là worksheet 1
48 End Sub
    
```

2.5.5. Điền mức so sánh (2)

Kiến thức tham khảo

- ✓ Lấy tọa độ điểm bất kỳ trên bản vẽ: mục 5.1.5 trang 217
- ✓ Tạo đối tượng văn bản: mục 5.2.6 trang 232.


Điểm bắt đầu vẽ được xác định dựa theo điểm người dùng chọn trên màn hình AutoCAD.

Biến MucSoSanh được gán giá trị đọc từ ô (1, 2) trong Sheet1 của file Excel.

```

50 Private Sub DienMucSoSanh()
51     'Xac dinh diem bat dau ve
52     ThisDrawing.Utility.Prompt ("Ban hay chon diem bat dau ve:")
53     DiemBatDauVe = ThisDrawing.Utility.GetPoint() 'Lay toa do diem nguoi dung Click
54
55     'Dien muc so sanh
56     MucSoSanh = WS.Cells(1, 2).Value 'Gan gia tri muc so sanh (1, 2)
57     ThisDrawing.ActiveLayer = ThisDrawing.Layers("ENTTDTUNHIEN")
58     SP = DiemBatDauVe
59     SP(0) = DiemBatDauVe(0) - 4 'Toa do X lui 4
60     SP(1) = DiemBatDauVe(1) + 0.5 'Toa do Y len 0.5
61     Set GhiChu = ThisDrawing.ModelSpace.AddText("MSS: " & MucSoSanh, SP, 0.5)
62 End Sub
    
```

▪ Kết quả



2.5.6. Vẽ đường mặt đất tự nhiên (3)

Kiến thức tham khảo

- ✓ Cấu trúc For ... To ...Next : mục 8.3.1 trang 46.
- ✓ Khai báo và vẽ đường đa tuyến rút gọn (LightWeightPolyline): mục 5.2.4 trang 228.

Số cọc cần vẽ được đếm thông qua câu lệnh lặp Do Until

Mỗi vòng lặp sẽ thực hiện thao tác tạo dữ liệu cho mảng tọa độ trắc dọc (3)

```

64 Private Sub VePolylineMatDat()
65     'Đếm số cọc trong file Excel
66     TongSoCoc = 0
67     i = 3 'Dữ liệu có từ hàng thứ 3
68     Do Until WS.Cells(i, 1).Value = "" 'Lặp đến khi giá trị = ""
69         i = i + 1 'Bỏ qua hàng tiêu đề
70         TongSoCoc = TongSoCoc + 1 'Cộng dồn số cọc
71     Loop
72     KCCD = DiemBatDauVe(0) 'Lấy tọa độ X của điểm bắt đầu vẽ
73     ReDim ToaDoTDTN(0 To TongSoCoc * 2 - 1) As Double 'Kích thước mảng tọa độ trắc dọc
74     'Lặp để vẽ đường đóng, ghi chú và tạo mảng Poyline2D tọa độ trắc dọc
75     For i = 1 To TongSoCoc
76         'Tính tọa độ các đỉnh của trắc dọc
77         KCCD = KCCD + WS.Cells(i + 2, 3) 'Tính tọa X
78         ToaDoTDTN(2 * i - 2) = KCCD 'Gán tọa độ X của đỉnh trắc dọc
79         ToaDoTDTN(2 * i - 1) = DiemBatDauVe(1) + (WS.Cells(i + 2, 2) - MucSoSanh) 'Tính tọa độ Y
80     Next
81
82     'Vẽ đường mặt đất tự nhiên
83     Call ThisDrawing.ModelSpace.AddLightWeightPolyline(ToaDoTDTN)
84 End sub
85
    
```

▪ Kết quả của phương thức trên:



2.5.7. Vẽ đường đóng đứng của trắc dọc (4)

★ Kiến thức tham khảo

Kiến thức tham khảo


- ✓ Thiết lập lớp hiện hành: mục 5.5.3 trang 262
- ✓ Khai báo và vẽ đoạn thẳng: mục 5.2.4 trang 226.

```

86 Private Sub VeDuongDongDung ()
87     For i = 1 To TongSoCoc
88         'Vẽ đường đóng đứng vùng trắc dọc
89         ThisDrawing.ActiveLayer = ThisDrawing.Layers ("LINEDONGTD")
90         SP (0) = ToaDoTDTN (2 * i - 2) 'Toạ độ X
91         SP (1) = ToaDoTDTN (2 * i - 1) 'Toạ độ Y
92         EP = SP: EP (1) = DiemBatDauVe (1)
93         Call ThisDrawing.ModelSpace.AddLine (SP, EP)
94         'Vẽ đường đóng đứng trắc dọc vùng ghi chú
95         ThisDrawing.ActiveLayer = ThisDrawing.Layers ("ENTTDTUNHIEN")
96         SP (1) = DiemBatDauVe (1) - 4.5 'Toạ độ Y xuống 4.5
97         EP (1) = DiemBatDauVe (1) - 6.5 'Toạ độ Y xuống 6.5
98         Call ThisDrawing.ModelSpace.AddLine (SP, EP)
99     Next
100 End Sub

```

- Kết quả



2.5.8. Điền thông số trắc dọc: cao độ tự nhiên (5), khoảng cách cộng dồn (6), khoảng cách lẻ (7) và tên cọc (8)


Kiến thức tham khảo

- ✓ Khai báo và sử dụng văn bản (Text): mục 5.2.6 trang 232.

```

102 Private Sub DienThongSoTracDoc
103 For i = 1 To TongSoCoc
104     'Điền cao độ tự nhiên
105     ThisDrawing.ActiveLayer = ThisDrawing.Layers("ENTTDTUNHIEN")
106     EP(0) = ToaDoTDTN(2 * i - 2) + 0.25 'Lệch phải 0.25
107     EP(1) = DiemBatDauVe(1) - 2 'Toa độ Y xuống 2
108     Set GhiChu = ThisDrawing.ModelSpace.AddText(WS.Cells(i + 2, 2), EP, 0.5)
109     GhiChu.Rotation = 3.14 / 2 'Góc quay 90o
110     'Điền khoảng cách cộng dồn
111     EP(1) = DiemBatDauVe(1) - 4 'Toa độ Y xuống 4
112     Set GhiChu = ThisDrawing.ModelSpace.AddText(ToaDoTDTN(2 * i - 2) - DiemBatDauVe(0), EP, 0.5)
113     GhiChu.Rotation = 3.14 / 2 'Góc quay 90o
114     'Điền khoảng cách lẻ
115     ThisDrawing.ActiveLayer = ThisDrawing.Layers("ENTTDTUNHIEN")
116     EP(0) = EP(0) + 0.25 - WS.Cells(i + 2, 3) / 2
117     EP(1) = DiemBatDauVe(1) - 6 'Toa độ Y xuống 6
118     KhoangCachLe = WS.Cells(i + 2, 3)
119     If WS.Cells(i + 2, 3) = 0 Then KhoangCachLe = "" 'Không điền nếu bằng 0
120     Set GhiChu = ThisDrawing.ModelSpace.AddText(KhoangCachLe, EP, 0.5)
121     'Điền tên cọc
122     ThisDrawing.ActiveLayer = ThisDrawing.Layers("TEXTTENCOC")
123     EP(0) = EP(0) - 0.5
124     EP(1) = DiemBatDauVe(1) - 8 'Toa độ Y xuống 8
125     Set GhiChu = ThisDrawing.ModelSpace.AddText(WS.Cells(i + 2, 1), EP, 0.5)
126 Next
127 AppEX.Quit 'Thoát ứng dụng Excel chạy ngầm vì không cần dùng nữa
128 End Sub
    
```

▪ Kết quả



2.5.9. Hoàn thiện trắc dọc: Vẽ đường đóng đúng phần ghi chú (9) - Vẽ đường đóng ngang phần ghi chú (10) - Ghi nội dung đầu trắc dọc (11)

★ Kiến thức tham khảo

Kiến thức tham khảo

- ✓ Thiết lập lớp hiện hành: mục 5.5.3 trang 262
- ✓ Khai báo và vẽ đoạn thẳng: mục 5.2.4 trang 226.
- ✓ Khai báo và sử dụng văn bản (Text): mục 5.2.6 trang 232.

```

130 Private Sub HoanThienTracDoc ()
131
132 End Sub
    
```

```

131 'Vẽ đường dóng ngang phần ghi chú
132     SP(0) = DiemBatDauVe(0) - 10
133     SP(1) = DiemBatDauVe(1)
134     EP = SP
135     EP(0) = ToaDoTDTN(UBound(ToaDoTDTN) - 1) + 1
136     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
137     SP(1) = SP(1) - 2.5
138     EP(1) = EP(1) - 2.5
139     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
140     SP(1) = SP(1) - 2
141     EP(1) = EP(1) - 2
142     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
143     SP(1) = SP(1) - 2
144     EP(1) = EP(1) - 2
145     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
146     SP(1) = SP(1) - 2
147     EP(1) = EP(1) - 2
148     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
149 'Vẽ đường dóng đứng phần ghi chú
150     SP(0) = ToaDoTDTN(UBound(ToaDoTDTN) - 1) + 1
151     SP(1) = DiemBatDauVe(1) - 8.5
152     EP = SP
153     EP(1) = DiemBatDauVe(1)
154     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
155     SP(0) = ToaDoTDTN(0) - 1
156     SP(1) = DiemBatDauVe(1) - 8.5
157     EP = SP
158     EP(1) = DiemBatDauVe(1)
159     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
160     SP(0) = ToaDoTDTN(0) - 10
161     SP(1) = DiemBatDauVe(1) - 8.5
162     EP = SP
163     EP(1) = DiemBatDauVe(1)
164     Call ThisDrawing.ModelSpace.AddLine(SP, EP)
165 'Nội dung đầu trắc dọc
166     ThisDrawing.ActiveLayer = ThisDrawing.Layers("TEXTHEADERTN")
167     SP(0) = DiemBatDauVe(0) - 9
168     SP(1) = DiemBatDauVe(1) - 2
169     Set GhiChu = ThisDrawing.ModelSpace.AddText("Cao do tu nhien", SP, 0.5)
170     SP(0) = DiemBatDauVe(0) - 9
171     SP(1) = DiemBatDauVe(1) - 4
172     Set GhiChu = ThisDrawing.ModelSpace.AddText("K/C cong don", SP, 0.5)
173     SP(0) = DiemBatDauVe(0) - 9
174     SP(1) = DiemBatDauVe(1) - 6
175     Set GhiChu = ThisDrawing.ModelSpace.AddText("Khoang cach le", SP, 0.5)
176     SP(0) = DiemBatDauVe(0) - 9
177     SP(1) = DiemBatDauVe(1) - 8
178     Set GhiChu = ThisDrawing.ModelSpace.AddText("Ten coc", SP, 0.5)
179 End Sub


```

▪ Kết quả


MSS 2														
Cao do tu nhien	0	0	2	0	0	0	0	0	0	0	0	0	0	0
K/C cong don	0	0	0	0	0	0	0	0	0	0	0	0	0	0
Khoang cach le	3	6	4	6	5	6	6	4	6	3	2	3		
Ten coc	KM01	2	3	4	TD1	P1	TC1	8	9	10	11	12		

2.6. Chạy thử chương trình và kiểm tra kết quả

- Trong giao diện AutoCAD bấm tổ hợp phím **Alt + F8** → Chọn “**VeTracDocTuNhien**” trong **Macro name** → Chọn **Run**
- Chọn vị trí muốn vẽ trắc ngang:



- Kết quả




3. BÀI TẬP

3.1. Bài tập 1

- Lập chương trình vẽ mặt cắt chữ T trong AutoCAD với dữ liệu của mặt cắt được lưu trữ trong file Excel.
- Dữ liệu trong Excel có định dạng như sau (chú ý đến vị trí của hàng và cột):

	A	B	C	D	E	F	G
1	ID	B1	B2	B3	H1	H2	H3
2	1	0.8	0.2	0.3	0.15	1	0.15
3	2	1	0.15	0.4	0.15	0.8	0.15
4	3	0.8	0.15	0.4	0.1	0.7	0.15
5							

- UserForm của chương trình được thiết kế như sau (VBA trong AutoCAD):



- Chương trình sẽ hoạt động như sau:
 - Người dùng nhập mã mặt cắt (theo các số trong cột A của file Excel chứa dữ liệu).
 - Khi chọn Nhập, dữ liệu sẽ được điền vào các TextBox tương ứng với B1, B2, B3
 - Thông báo lỗi nếu mã mặt cắt không có trong file dữ liệu.
 - Khi chọn Vẽ hình, chương trình sẽ yêu cầu người dùng chọn một điểm trên bản vẽ AutoCAD và thực hiện vẽ mặt cắt chữ T theo các thông số tương ứng trong các TextBox chứa giá trị của B1, B2, ...
 - Mặt cắt được vẽ trong AutoCAD với đối tượng LightweightPolyline

3.2. Bài tập 2

- Lập chương trình tính và vẽ biểu đồ mô men của một dầm giản đơn.
- Tải trọng tác dụng: tải trọng phân bố đều và 01 tải trọng tập trung ở vị trí bất kỳ.

- Chương trình sẽ hoạt động như sau:
 - Người dùng nhập chiều dài dầm, giá trị của tải trọng phân bố đều, giá trị của tải trọng tập trung, khoảng cách từ đầu dầm bên trái đến vị trí đặt tải trọng tập trung.
 - Chương trình sẽ tính và vẽ biểu đồ mô men của kết cấu (do 2 loại tải trọng trên gây ra) trong AutoCAD.
- Hướng dẫn cơ bản:
 - Khởi động VBA IDE trong AutoCAD.
 - Tạo ra một Userform và thiết kế giao diện. Sẽ cần có 4 TextBox nhập dữ liệu (chiều dài dầm, độ lớn của tải trọng phân bố đều, độ lớn của tải trọng tập trung, khoảng cách từ đầu dầm bên trái đến vị trí của tải trọng tập trung). Cần có 2 Button để điều khiển hoạt động của chương trình (một Button để tính và vẽ, một Button để thoát khỏi chương trình).
 - Viết mã lệnh tính toán nội lực (mô men) cho khoảng 100 mặt cắt (cách đều nhau) của dầm.
 - Vẽ biểu đồ mô men do tải trọng gây ra, sử dụng đối tượng **Line** hoặc **LWPolyline** để vẽ trong AutoCAD.

DỰ ÁN V: CHƯƠNG TRÌNH CHUYỂN ĐỔI SỐ LIỆU TỪ BẢN VẼ KHẢO SÁT ĐỊA HÌNH TRÊN AUTOCAD SANG SỐ LIỆU DẠNG TỌA ĐỘ ĐIỂM TRÊN EXCEL ĐỂ LÀM DỮ LIỆU CHO CÁC PHẦN MỀM THIẾT KẾ ĐƯỜNG

1. NỘI DUNG DỰ ÁN

1.1. Mô tả yêu cầu

- Tên dự án: Chương trình chuyển đổi số liệu từ bản vẽ khảo sát địa hình trên AutoCAD sang số liệu dạng tọa độ điểm trên Excel để làm dữ liệu cho các phần mềm thiết kế đường.
- Dữ liệu ban đầu: Bản vẽ AutoCAD (DWG) chứa số liệu khảo sát của một khu vực. Số liệu bao gồm các điểm đo được ghi theo quy ước trắc địa trên bản vẽ và các ký hiệu khác (tham khảo bản vẽ KS1.DWG). Số liệu trên bản vẽ này không thể sử dụng trực tiếp làm dữ liệu đầu vào cho những phần mềm dựng bản đồ số, vì chúng là các đối tượng của AutoCAD chứ không phải là dữ liệu mà các phần mềm thiết kế có thể nhận dạng.



- Hầu hết các phần mềm thiết kế đường đều có mô-đun xây dựng bản đồ số, mô-đun này chấp nhận dữ liệu đo toàn đạc từ file Text với cấu trúc phổ biến là (Tên điểm đo, X, Y, Z, Ghi chú) chứ không nhận biết một cách trực tiếp dữ liệu có trong bản vẽ như file KS1.DWG. Trong khi đó, file KS1.DWG là dạng dữ liệu mà các đơn vị khảo sát cung cấp, nó chỉ phù hợp cho kỹ sư đọc, chứ không phù hợp với các phần mềm thiết kế đường. Do đó cần chuyển đổi dữ liệu về dạng khác, phù hợp làm dữ liệu đầu vào cho phần mềm thiết kế đường (như Nova-TDN, Civil 3D). Trong dự án này sẽ cần chuyển đổi dữ liệu từ bản vẽ khảo sát địa hình trên AutoCAD thành dạng điểm đo có cấu trúc như sau: (Tên điểm đo, X, Y, Z). Do trong bản vẽ, các điểm đo không có ghi chú và tên nên thứ tự của điểm đo được lấy làm tên của điểm đo.

	A	B	C	D
1	0	803.4423	96.45935	5.75
2	1	811.6863	97.03731	5.79
3	2	809.6478	104.5823	5.7
4	3	813.2889	113.0077	5.69
5	4	797.4425	145.7885	5.55
6	5	788.9407	143.731	5.51
7	6	780.2269	135.662	5.62
8	7	795.5989	132.3352	5.67
9	8	792.858	140.558	5.62
10	9	779.2672	138.2549	5.63
11	10	789.6056	150.6715	5.6
12	11	679.5157	115.2151	5.76
13	12	680.3316	130.5246	5.84

- Dữ liệu trên bản vẽ có thể sai do người vẽ vô tình tạo ra (ví dụ cao độ đúng là 6.31 nhưng khi thực hiện vẽ thì đưa vào là 63.1). Do đó cần thực hiện việc kiểm tra sơ bộ cao độ các điểm đo (sau khi đã chuyển đổi sang Excel) để loại bỏ những nhầm lẫn kiểu này.

1.2. Cấu trúc dự án

1.2.1. Dữ liệu ban đầu


- File “KS1.dwg”: là một bản vẽ khảo sát địa hình, có tọa độ và cao độ của các điểm đo cũng như các ký hiệu địa hình địa vật khác. Cao độ của các điểm đo là đối tượng kiểu Text trên bản vẽ, còn tọa độ của đối tượng Text này chính là tọa độ của điểm đo. Tỷ lệ trên bản vẽ là 1/1000, tỷ lệ này hay được sử dụng nhất và cũng phù hợp để sử dụng trực tiếp cho các phần mềm thiết kế đường.




- File KS2.dwg cũng là một bản vẽ khảo sát địa hình khác dùng để thử nghiệm chương trình.

1.2.2. Giao diện người dùng

- Chương trình được sử dụng dưới dạng Macro của AutoCAD (xem mục 3.3 trang 196).




- Người dùng chọn vùng dữ liệu cần chuyển đổi theo cách chọn đối tượng thông thường của AutoCAD. Có thể chọn toàn bộ bằng lệnh All hoặc chọn một vùng bất kỳ hoặc bấm chọn từng đối tượng.



1.2.3. Khối chương trình (Module)

- Chương trình được đặt trong 01 Module (Module1).



- Khối mã lệnh định nghĩa kiểu dữ liệu người dùng (điểm đo).

- Khởi mã lệnh thực thi chức năng của chương trình được bố trí trong 1 chương trình con (chương trình con dạng sub có tên là Convert).
- Khởi mã lệnh chương trình kiểm tra sơ bộ dữ liệu trên Excel.

1.3. Nội dung kiến thức VBA liên quan đến dự án

- Kiến thức đã nêu trong các dự án trước
 - Yêu cầu khai báo biến trước khi sử dụng: mục 2 trang 23.
 - Định nghĩa và sử dụng kiểu dữ liệu người dùng: mục 5.8 trang 31.
 - Chương trình con: mục 9 trang 50 và mục 3.2.2 trang 113
 - Khai báo biến cơ bản: mục 5 trang 26 và mục 6 trang 33.
 - Khai báo và sử dụng mảng động: mục 5.4 trang 28.
 - Cấu trúc **For Each ... In ...** : mục 8.3.2 trang 48.
 - Cấu trúc **If ... Then ... Else ... End If** : mục 8.1 trang 42.
 - Cấu trúc **For ... To ...** : mục 8.3.1 trang 46.
 - Dữ liệu kiểu **Range**: dùng khi cần lưu trữ dữ liệu của 1 hoặc nhiều Cells của Excel (mục 5.2.5 trang 135).
 - **Application.InputBox**: dùng để tạo ra giao diện nhập dữ liệu đơn giản cho người dùng khi cần chọn một vùng các Cells trong Excel (mục 8.2.1 trang 165).
 - Đọc và ghi dữ liệu với **Columns()**, **Cells()**: mục 5.2.5 trang 135
 - **Lbound** và **Ubound**: hàm xác định chỉ số nhỏ nhất và lớn nhất của mảng (mục 9.3.5 trang 54).
 - Ghi dữ liệu vào Excel theo địa chỉ thay đổi theo biến: **Cells(row, col)** (mục 5.2.6 trang 140).
- Làm việc với đối tượng Text: mục 5.2.6 trang 232 và mục 5.4.4 trang 258.
- Làm việc với tập đối tượng SelectionSet: mục 5.3 trang 233.
- Bẫy lỗi với **On Error Resume Next**: mục 5.3 trang 233 (ứng dụng với SelectionSet) và mục 14.3 trang 94.
- Hiệu chỉnh đối tượng hình học (Text): mục 5.4 trang 244


1.4. Những lưu ý khác

- Phần hướng dẫn này phù hợp với phiên bản AutoCAD từ 2007 trở lên. Nếu sử dụng phiên bản AutoCAD 2010 trở lên cần cài đặt thêm VBA (chi tiết xem trong tài liệu hướng dẫn của Autodesk).
- Tài liệu tham khảo (mục, trang) được lấy theo Giáo trình môn Tự động hóa Thiết kế cầu đường (bản PDF).
- Dự án này sẽ dùng hình thức dự án độc lập (xem mục 2.1 trang 190).

2. TRÌNH TỰ THỰC HIỆN DỰ ÁN


2.1. Mở file KS1.dwg

- File dữ liệu ban đầu (KS1.dwg) là file bản vẽ khảo sát địa hình của một dự án xây dựng công trình giao thông trong thực tế. Số liệu khảo sát này do một đơn vị khảo sát thực hiện và họ chỉ chuyên giao tài liệu dưới dạng bản vẽ như trong file này.
- Sử dụng AutoCAD 2007 (hoặc phiên bản AutoCAD mới hơn) để mở file KS1.dwg



2.2. Chạy VBA IDE và tạo Module

- Gọi VBA IDE: bấm **Alt + F11**
- Tạo Module: Chọn Insert → Module



2.3. Xây dựng giao diện người sử dụng

- Chương trình này không có giao diện dạng UserForm mà thực hiện thông qua cách gọi Macro của AutoCAD và chế độ dòng lệnh của AutoCAD.

2.4. Xây dựng khối chương trình dạng khung

Thêm một Module vào dự án với tên được đặt là **Chuongtrinh**.

2.4.1. Viết mã lệnh định nghĩa kiểu dữ liệu người dùng cho **Điểm đo**

TT	Ý nghĩa	Tên biến	Kiểu dữ liệu	Ví dụ
1	Thứ tự điểm đo	TT	Long	123
2	Tọa độ theo phương X	X	Double	12.56
3	Tọa độ theo phương Y	Y	Double	52.36
4	Tọa độ theo phương Z (cao độ)	Z	Double	6.35

```

1 Option Explicit
2 Type DiemDo
3     TT As Long
4     X As Double
5     Y As Double
6     Z As Double
7 End Type
8
    
```

★ Kiến thức tham khảo

- ✓ Cách đặt tên biến: mục 1 trang 23
- ✓ Kiểu số: mục 5.2 và 5.3 trang 27.
- ✓ Khai báo và sử dụng kiểu dữ liệu tự định nghĩa: mục 5.8, trang 31.

2.4.2. Tạo chương trình con trong Module

- Chương trình con này sẽ thực hiện tất cả các nội dung của dự án và là dạng Macro của AutoCAD.

```

9 ' Lay toa do cua cac diem do tren ban ve cu
10 Public Sub Convert()
11
12 End Sub
    
```

★ Kiến thức tham khảo

- ✓ Từ khóa **Public**: mục 9 trang 50 (tham khảo thêm mục 6 trang 33).
- ✓ Chương trình con dạng thủ tục (**Sub**): mục 9.2, trang 51.
- ✓ Macro trong AutoCAD: mục 3 trang 194.

2.5. Viết mã lệnh chi tiết cho chương trình chuyển đổi dữ liệu

2.5.1. Định nghĩa các biến chính của chương trình

- Mã lệnh khai báo các biến chính của chương trình:

```

11 Dim DD() As DiemDo ' Mang dong chua cac diem do
12 Dim TextObj As AcadText ' Doi tuong Text cua AutoCAD
13 Dim SsetObj As AcadSelectionSet ' Tap doi tuong SelectionSet de lam viec voi cac doi tuong cua AutoCAD
14 Dim Elev As Double ' Cao do cua diem do
15 Dim Order As Long ' thu tu cua diem do
16 Dim InsPoint As Variant ' Toa do x, y cua diem do
17 Dim N As Long ' So doi tuong duoc chon trong ban ve AutoCAD
18 Dim i As Long ' Duyet doi tuong
19

```

- Do số lượng điểm đo là không thể xác định từ trước nên danh sách các điểm đo không thể cố định, vì thế phải sử dụng mảng động.
- Các điểm đo được chọn từ bản vẽ AutoCAD nên cần đến đối tượng SelectionSet (tên đầy đủ là AcadSelectionSet) để chứa tạm thời các Text được chọn.
- Tọa độ (X, Y) của điểm đo trên bản vẽ là một điểm (point) gồm 2 thành phần là X và Y. Ở đây sử dụng biến kiểu Variant để chứa point này, tuy nhiên có nhiều cách khác để thực hiện nhiệm vụ này.

★ Kiến thức tham khảo

- ✓ Khai báo biến cơ bản: mục 6.2 trang 37.
- ✓ Khai báo biến kiểu tự định nghĩa: mục 5.8 trang 31 và mục 6.3 trang 37.
- ✓ Khai báo mảng động: mục 6.5 trang 38
- ✓ Dữ liệu kiểu Variant: mục 5.7 trang 30
- ✓ Khai báo kiểu đối tượng hình học của AutoCAD: mục 5.2.2 trang 224 và mục 5.2.6 trang 232 về khai báo đối tượng AcadText
- ✓ Khai báo kiểu đối tượng SelectionSet: mục 5.3.1 trang 234

2.5.2. Loại bỏ đối tượng SelectionSet cũ

- AutoCAD có thể chứa cùng lúc nhiều đối tượng SelectionSet.
- Đối tượng SelectionSet dùng để chứa các Text (các điểm đo) dự kiến đặt tên là MySset, tuy nhiên việc đặt tên này sẽ bị lỗi khiến cho chương trình không thể chạy được khi trong AutoCAD (vì một lý do nào đó) đã có sẵn một tập đối tượng có tên như vậy.
- Có vài cách thức khác nhau để tránh được lỗi này (tham khảo mục 5.3.1 trang 234)
- Đoạn mã lệnh sau sẽ loại bỏ đối tượng SelectionSet có tên là **MySset** (nếu có):

```

19
20 'Tim trong ban ve AutoCAD xem da co tap doi tuong MySset nao khong, neu co thi xoa no di
21 For Each SsetObj In ThisDrawing.SelectionSets
22     If SsetObj.Name = "MySset" Then
23         SsetObj.Delete
24         Exit For
25     End If
26 Next
27

```

★ Kiến thức tham khảo

- ✓ Đối tượng SelectionSet: mục 5.3 trang 233
- ✓ Đối tượng ThisDrawing: mục 4.2.2 trang 202

2.5.3. Khởi tạo đối tượng SelectionSet

- Thêm vào một đối tượng SelectionSet có tên là MySset dùng để lưu trữ tạm thời các Text (điểm đo) khi thực hiện việc lựa chọn (select) trên bản vẽ.


```

27
28      ' Khởi tạo tập đối tượng mới để chứa các đối tượng AutoCAD
29      Set SsetObj = ThisDrawing.SelectionSets.Add("MySSet")
30

```

★ **Kiến thức tham khảo**

✓ Khởi tạo đối tượng SelectionSet: mục 5.3.1 trang 234

2.5.4. Nhập dữ liệu

- Dữ liệu được nhập vào bằng hình thức chọn đối tượng thông dụng của AutoCAD. Giao diện là lời nhắc (prompt) của AutoCAD sẽ xuất hiện trên màn hình (phiên bản từ AutoCAD 2007 được cài mặc định) hoặc trên cửa sổ dòng lệnh (Command Window) đối với các phiên bản AutoCAD thấp hơn hoặc có những thiết lập riêng (xem trong tài liệu hướng dẫn sử dụng AutoCAD, lệnh Options).
- Mã lệnh:

```

30
31      ' Dùng Mouse để chọn các điểm đo cần chuyển đổi
32      SsetObj.SelectOnScreen
33

```

- Việc nhập dữ liệu sẽ kết thúc khi người dùng bấm Enter.
- Câu lệnh trên cho phép chọn mọi loại đối tượng (Line, Pline, Circle, **Text**, ...) mà bạn có thể nhìn thấy được trong bản vẽ (đối tượng hình học).

★ **Kiến thức tham khảo**

✓ Các phương thức của đối tượng SelectionSet: mục 5.3.2 trang 235

2.5.5. Lấy số đối tượng đã được chọn

- Sau khi người dùng bấm phím Enter thì việc lựa chọn kết thúc, trong đối tượng SelectionSet (tên là MySSet) đã chứa danh sách các đối tượng được chọn.
- Trong số các đối tượng được chọn trong danh sách trên có thể có những đối tượng không phải là Text. Do đó cần duyệt toàn bộ danh sách và chỉ thực hiện chuyển đổi với những đối tượng Text.
- Mã lệnh:

```

33
34      ' Lấy số đối tượng đã chọn được
35      N = SsetObj.Count
36

```

★ **Kiến thức tham khảo**

✓ Các phương thức của đối tượng SelectionSet: mục 5.3.2 trang 235

2.5.6. Lập danh sách các điểm đo

- Danh sách các điểm đo chứa trong mảng động đã khai báo ở trên dưới dạng một mảng động (mảng chưa biết có bao nhiêu phần tử, chỉ biết tên và kiểu dữ liệu khi khai báo biến).
- Chỉ số đầu tiên của danh sách là 0 (gán ở dòng lệnh 38).

- Thực hiện lệnh **For ... To ...** để duyệt toàn bộ các đối tượng được chọn có trong đối tượng SelectionSet (tên là MySSet). Sẽ có 2 trường hợp xảy ra khi duyệt từng đối tượng được chọn:
 - Đối tượng được chọn không phải là Text.
 - Đối tượng được chọn là Text (và được coi là dữ liệu điểm đo trong chương trình này)
- Mã lệnh duyệt từng đối tượng trong MySSet:

```

36
37     ' Khoi tao so thu tu cua cac diem do
38     Order = 0
39     For i = 0 To (N - 1)
40
41     Next i

```

- Kiểm tra xem từng đối tượng được chọn trong MySSet có phải là Text không:
 - Nếu không phải thì không thực hiện việc chuyển đổi dữ liệu (không làm gì cả!)
 - Nếu là Text thì lấy thông tin về tọa độ (X,Y) và cao độ (Z).

```

39     For i = 0 To (N - 1)
40     If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41
42     End If
43     Next i

```

- ✓ Lấy thông tin về đối tượng chứa trong đối tượng SelectionSet: mục 5.3.2 trang 235
- ✓ Loại đối tượng trong AutoCAD thường có tên bắt đầu bằng **AcDb**

- Nếu đối tượng được chọn **thứ i** trong đối tượng MySSet là Text thì:
 - Gán nó cho biến đối tượng kiểu Text để tiện cho việc xử lý (dòng lệnh 41).

```

40     If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41     Set TextObj = SsetObj.Item(i)
42     ' Bay loi de loai bo cac doi tuong Text khong phai la so
43     On Error Resume Next
44     Elev = TextObj.TextString ' Lay noi dung cua Text lam cao do diem do
45     If Err.Number = 0 Then ' Neu doi tuong Text la so
46     ReDim Preserve DD(Order) ' Cap phat bo nho cho mang dong
47     DD(Order).TT = Order
48     TextObj.Color = acGreen ' Danh dau diem do duoc chuyen doi trong ban ve AutoCAD
49     InsPoint = TextObj.InsertionPoint ' Lay toa do X, Y cho diem do
50     DD(Order).X = InsPoint(0)
51     DD(Order).Y = InsPoint(1)
52     DD(Order).Z = Elev
53     Order = Order + 1
54     Else ' Neu doi tuong Text khong phai la so
55     Err.Clear
56     End If
57     End If

```

- Thực hiện bẫy lỗi do khả năng chọn phải đối tượng Text nhưng không phải là điểm đo trên bản vẽ (ví dụ chọn nhầm vào phần ghi chú địa hình: **Nhà cấp 4**) nên giá trị của Text không thể chuyển sang dạng số được (Text có giá trị "**Nhà cấp 4**" không thể chuyển sang dạng số được). Điều này sẽ gây lỗi làm chương trình bị dừng lại, do đó cần loại trừ lỗi này bằng cách bẫy lỗi.

- ✓ Bẫy lỗi: mục 14.3 trang 94

- Bật chế độ bắt lỗi (dòng lệnh 43) trước khi thực hiện việc lấy cao độ của điểm đo (dòng lệnh 44):

```

40 If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41   Set TextObj = SsetObj.Item(i)
42   ' Bay loi de loai bo cac doi tuong Text khong phai la so
43   On Error Resume Next
44   Elev = TextObj.TextString ' Lay noi dung cua Text lam cao do diem do
45   If Err.Number = 0 Then ' Neu doi tuong Text la so
46     ReDim Preserve DD(Order) ' Cap phat bo nho cho mang dong
47     DD(Order).TT = Order
48     TextObj.Color = acGreen ' Danh dau diem do duoc chuyen doi trong ban ve AutoCAD
49     InsPoint = TextObj.InsertionPoint ' Lay toa do X, Y cho diem do
50     DD(Order).X = InsPoint(0)
51     DD(Order).Y = InsPoint(1)
52     DD(Order).Z = Elev
53     Order = Order + 1
54   Else ' Neu doi tuong Text khong phai la so
55     Err.Clear
56   End If
57 End If

```

- Nếu lỗi xảy ra (gặp phải tình huống: Text không phải điểm đo) thì do đã bật chế độ bắt lỗi nên chương trình sẽ không bị dừng lại, mà nó tiếp tục chạy với giá trị của biến Elev không xác định, đồng thời đối tượng Err (xem mục 14.3.2 trang 95) sẽ chứa mã lỗi. Trong trường hợp gặp lỗi này, do Text không phải là điểm đo nên sẽ không thực hiện việc chuyển đổi (tức là không gán giá trị cho mảng chứa danh sách các điểm đo) mà chỉ thực hiện xóa bỏ việc bắt lỗi đối với vòng lặp thứ i này (vòng lặp thứ i+1 sẽ tiếp tục đi qua dòng lệnh 43 nên việc bắt lỗi sẽ được thiết lập lại)

```

40 If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41   Set TextObj = SsetObj.Item(i)
42   ' Bay loi de loai bo cac doi tuong Text khong phai la so
43   On Error Resume Next
44   Elev = TextObj.TextString ' Lay noi dung cua Text lam cao do diem do
45   If Err.Number = 0 Then ' Neu doi tuong Text la so
46     ReDim Preserve DD(Order) ' Cap phat bo nho cho mang dong
47     DD(Order).TT = Order
48     TextObj.Color = acGreen ' Danh dau diem do duoc chuyen doi trong ban ve AutoCAD
49     InsPoint = TextObj.InsertionPoint ' Lay toa do X, Y cho diem do
50     DD(Order).X = InsPoint(0)
51     DD(Order).Y = InsPoint(1)
52     DD(Order).Z = Elev
53     Order = Order + 1
54   Else ' Neu doi tuong Text khong phai la so
55     Err.Clear
56   End If
57 End If

```

- Nếu không xảy ra lỗi (giả định đó là điểm đo), lúc này giá trị Err.Number = 0, thì thực hiện việc chuyển đổi dữ liệu như sau:
 - Cấp phát mảng động (dòng lệnh 46).
 - Cao độ của điểm đo (Z) được lấy bằng giá trị của biến Elev tại dòng lệnh 52 (biến này nhận giá trị của Text, được chuyển đổi ở dòng lệnh 44).
 - Thứ tự của điểm đo (cũng được xem là tên của điểm đo) được lấy bằng thứ tự của nó trong danh sách (biến Order, dòng lệnh 47).
 - Tọa độ (X, Y) của điểm đo lấy bằng tọa độ điểm chèn của Text (dòng lệnh 49→51).
 - Đổi màu Text được coi là điểm đo sang màu Green (dòng lệnh 48)

- Tăng chỉ số của mảng động lên 1 (dòng lệnh 53)

```

40 If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41   Set TextObj = SsetObj.Item(i)
42   ' Bay loi de loai bo cac doi tuong Text khong phai la so
43   On Error Resume Next
44   Elev = TextObj.TextString ' Lay noi dung cua Text lam cao do diem do
45   If Err.Number = 0 Then ' Neu doi tuong Text la so
46     ReDim Preserve DD(Order) ' Cap phat bo nho cho mang dong
47     DD(Order).TT = Order
48     TextObj.Color = acGreen ' Danh dau diem do duoc chuyen doi trong ban ve AutoCAD
49     InsPoint = TextObj.InsertionPoint ' Lay toa do X, Y cho diem do
50     DD(Order).X = InsPoint(0)
51     DD(Order).Y = InsPoint(1)
52     DD(Order).Z = Elev
53     Order = Order + 1
54   Else ' Neu doi tuong Text khong phai la so
55     Err.Clear
56   End If
57 End If

```

- ✓ Hiệu chỉnh đối tượng Text: mục 5.4.4 trang 258
- ✓ Sử dụng các phương thức và thuộc tính của đối tượng hình học: mục 5.4.1 trang 245 và mục 5.4.2 trang 252
- ✓ Sử dụng đối tượng Err: mục 14.3.2 trang 95

2.5.7. Xóa bỏ đối tượng MySset sau khi sử dụng

- Việc xóa đối tượng này sẽ không gây mất dữ liệu trên bản vẽ vì nó chỉ lưu trữ danh sách các đối tượng được chọn.

```

59 ' Xoa bo tap doi tuong sau khi dung
60 SsetObj.Delete
61

```

2.5.8. Kết nối với Excel và xuất dữ liệu sau khi chuyển đổi

- Chạy Excel (dòng lệnh 63 → 65) → Tạo mới một file Excel (dòng lệnh 67) → Tạo mới sheet có tên là "SoLieu" (dòng lệnh 68, 69):

```

61
62 ' Ket noi voi Excel
63 Dim App As Excel.Application
64 Set App = Excel.Application
65 App.Visible = True
66 Dim WBook As Workbook, WSheet As Worksheet
67 Set WBook = App.Workbooks.Add
68 Set WSheet = WBook.Worksheets(1)
69 WSheet.Name = "SoLieu"
70

```

- ✓ Làm việc với Workbook: mục 7.2 trang 154
- ✓ Làm việc với Worksheet: mục 7.3 trang 155

- Ghi dữ liệu từ mảng chứa thông tin về điểm đo ra Excel:

```
70
71 For i = LBound(DD) To UBound(DD)
72     WBook.Worksheets("SoLieu").Cells(i + 1, 1) = DD(i).TT
73     WBook.Worksheets("SoLieu").Cells(i + 1, 2) = DD(i).X
74     WBook.Worksheets("SoLieu").Cells(i + 1, 3) = DD(i).Y
75     WBook.Worksheets("SoLieu").Cells(i + 1, 4) = DD(i).Z
76 Next i
77
78 App.Caption = "Ket Thuc"
79 End Sub
```

✓ Làm việc với Cells: mục 7.4 trang 156

2.5.9. Mã lệnh của toàn bộ chương trình

```

1 Option Explicit
2 Type DiemDo
3     TT As Long
4     X As Double
5     Y As Double
6     Z As Double
7 End Type
8
9 ' Lay toa do cua cac diem do tren ban ve cu de chuyen thanh du lieu thiet ke cho Nova-TDN và Civil 3D
10 Public Sub Convert()
11     Dim DD() As DiemDo ' Mang dong chua cac diem do
12     Dim TextObj As AcadText ' Doi tuong Text cua AutoCAD
13     Dim SsetObj As AcadSelectionSet ' Tap doi tuong SelectionSet de lam viec voi cac doi tuong cua AutoCAD
14     Dim Elev As Double ' Cao do cua diem do
15     Dim Order As Long ' thu tu cua diem do
16     Dim InsPoint As Variant ' Toa do x, y cua diem do
17     Dim N As Long ' So doi tuong duoc chon trong ban ve AutoCAD
18     Dim i As Long ' Duyet doi tuong
19
20     'Tim trong ban ve AutoCAD xem da co tap doi tuong MySset nao khong, neu co thi xoa no di
21     For Each SsetObj In ThisDrawing.SelectionSets
22         If SsetObj.Name = "MySset" Then
23             SsetObj.Delete
24             Exit For
25         End If
26     Next
27
28     ' Khoi tao tap doi tuong moi de chua cac doi tuong AutoCAD
29     Set SsetObj = ThisDrawing.SelectionSets.Add("MySset")
30
31     ' Dung Mouse de chon cac diem do can chuyen doi
32     SsetObj.SelectOnScreen
33
34     ' Lay so doi tuong da chon duoc
35     N = SsetObj.Count
36
37     ' Khoi tao so thu tu cua cac diem do
38     Order = 0
39     For i = 0 To (N - 1)
40         If SsetObj.Item(i).ObjectName = "AcDbText" Then ' Chi lam viec voi cac doi tuong dang TEXT
41             Set TextObj = SsetObj.Item(i)


```

```
42      ' Bay loi de loai bo cac doi tuong Text khong phai la so
43      On Error Resume Next
44      Elev = TextObj.TextString ' Lay noi dung cua Text lam cao do diem do
45      If Err.Number = 0 Then ' Neu doi tuong Text la so
46          ReDim Preserve DD(Order) ' Cap phat bo nho cho mang dong
47          DD(Order).TT = Order
48          TextObj.Color = acGreen ' Danh dau diem do duoc chuyen doi trong ban ve AutoCAD
49          InsPoint = TextObj.InsertionPoint ' Lay toa do X, Y cho diem do
50          DD(Order).X = InsPoint(0)
51          DD(Order).Y = InsPoint(1)
52          DD(Order).Z = Elev
53          Order = Order + 1
54      Else ' Neu doi tuong Text khong phai la so
55          Err.Clear
56      End If
57  End If
58  Next i
59  ' Xoa bo tap doi tuong sau khi dung
60  SsetObj.Delete
61
62  ' Ket noi voi Excel
63  Dim App As Excel.Application
64  Set App = Excel.Application
65  App.Visible = True
66  Dim WBook As Workbook, WSheet As Worksheet
67  Set WBook = App.Workbooks.Add
68  Set WSheet = WBook.Worksheets(1)
69  WSheet.Name = "DuLieu"
70
71  For i = LBound(DD) To UBound(DD)
72      WBook.Worksheets("DuLieu").Cells(i + 1, 1) = DD(i).TT
73      WBook.Worksheets("DuLieu").Cells(i + 1, 2) = DD(i).X
74      WBook.Worksheets("DuLieu").Cells(i + 1, 3) = DD(i).Y
75      WBook.Worksheets("DuLieu").Cells(i + 1, 4) = DD(i).Z
76  Next i
77
78  App.Caption = "Ket Thuc"
79 End Sub
```

2.6. Chạy thử chương trình chuyển đổi dữ liệu

2.6.1. Chạy thử chương trình

- Đặt con trỏ bên trong chương trình con **Convert** → Bấm chọn nút **Run** (như hình dưới):



- Chọn các đối tượng, có thể chọn bất kỳ loại đối tượng nào hoặc toàn bộ đối tượng hiện có trong bản vẽ (nếu như cần chuyển đổi toàn bộ các điểm đo có trong bản vẽ) → Bấm Enter để kết thúc việc chọn đối tượng chuyển đổi:



- Kết quả: chương trình sẽ tự động chạy Excel và chuyển kết quả sang, cột A là thứ tự (hay tên điểm đo), cột B, C, D là tọa độ X, Y, Z:

	A	B	C	D	E
1	1	803.4423	96.45935	5.75	
2	2	811.6863	97.03731	5.79	
3	3	809.6478	104.5823	5.7	
4	4	813.2889	113.0077	5.69	
5	5	797.4425	145.7885	5.55	
6	6	788.9407	143.731	5.51	
7	7	780.2269	135.662	5.62	
8	8	795.5989	132.3352	5.67	
9	9	792.858	140.558	5.62	
10	10	779.2672	138.2549	5.63	
11	11	789.6056	150.6715	5.6	
12	12	679.5157	115.2151	5.76	
13	13	680.3316	130.5246	5.84	


2.7. Viết mã lệnh cho chương trình kiểm tra sơ bộ dữ liệu trong Excel

- Kết quả sau khi được chuyển sang Excel, cần kiểm tra lại một cách sơ bộ để phát hiện những lỗi do người vẽ bản vẽ khảo sát vô tình gây ra, ví dụ như nhập dữ liệu sai.

- Giả định đã biết cao độ của vùng này không quá 10.00m, do đó sẽ tiến hành kiểm tra xem có điểm đo nào có cao độ quá 10m không, nếu có thì đánh dấu bằng cách đổi màu chữ của điểm đó thành màu **đỏ đậm**.

2.7.1. Tạo mô-đun để viết mã lệnh

- Trong file kết quả do chương trình chuyển đổi xuất sang Excel → Gọi VBA IDE (có thể bấm **Alt+F11**) → Tạo **Module1**:



2.7.2. Tạo chương trình con

- Chương trình con dạng Sub:

```

1 Option Explicit
2
3 Public Sub Test ()
4
5 End Sub

```

2.7.3. Mã lệnh kiểm tra cao độ

- Chương trình sẽ tự động xác định vùng có dữ liệu trong Excel bằng cách sử dụng đối tượng UsedRange.
- Do chỉ quan tâm đến cao độ của các điểm đo, nên chỉ cần duyệt các Cells trong cột D (tức là cột số 4):

```

4 Dim r As Range
5 For Each r In ThisWorkbook.Worksheets("SoLieu").UsedRange.Columns(4).Cells
6     If r.Value > 10 Then
7         r.Font.Color = vbRed
8         r.Font.Bold = True
9     End if
10 Next

```

- Biến **r** (kiểu Range) sẽ chứa thông tin đọc được trong các Cells thuộc cột 4 (là cột chứa giá trị cao độ của điểm đo). Kiểm tra giá trị trong biến **r** (r.Value), nếu nó lớn hơn 10 thì thực hiện việc đổi chữ sang màu **đỏ đậm**.

```

4 Dim r As Range
5 For Each r In ThisWorkbook.Worksheets("SoLieu").UsedRange.Columns(4).Cells
6     If r.Value > 10 Then
7         r.Font.Color = vbRed
8         r.Font.Bold = True
9     End if
10 Next
    
```

- ✓ Đối tượng UsedRange: mục 7.4.3 trang 157
- ✓ Thay đổi định dạng trong Cells: mục 2.2.2 trang 108

3. BÀI TẬP

3.1. Bài tập 1

- Lập chương trình tính tổng chiều dài của các đoạn thẳng (đối tượng kiểu Line) được chọn trên bản vẽ AutoCAD.
- Chương trình sẽ thực hiện như sau:
 - Yêu cầu người dùng chọn các đối tượng trên bản vẽ. Có thể chọn từng đối tượng, chọn một nhóm hoặc chọn toàn bộ.
 - Lọc lấy những đối tượng kiểu **AcDbLine** và tính tổng chiều dài của những đối tượng này.
 - Yêu cầu người dùng chọn 1 điểm (point) trên bản vẽ và ghi kết quả (tổng chiều dài các đoạn thẳng được chọn) lên bản vẽ tại điểm (point) vừa chọn.
- Hướng dẫn cơ bản:
 - Tên đối tượng hình học của AutoCAD được chứa trong thuộc tính ObjectName. Đoạn chương trình sau sẽ hiển thị tên của mọi loại đối tượng hình học trong cửa sổ Immediate khi được chọn:

```

Public Sub TenDT()
    Dim Obj As AcadObject
    Dim Pnt As Variant

    ThisDrawing.Utility.GetEntity Obj, Pnt, "Chon doi tuong:"

    Debug.Print Obj.ObjectName
End Sub
    
```

- Tên một số đối tượng hình học hay sử dụng:

Loại đối tượng	Tên
Line	AcDbLine
LWPolyLine	AcDbPolyline
Text	AcDbText

- Chiều dài của đối tượng dạng đường (Line, LWPolyLine) được chứa trong thuộc tính Length. Đoạn mã lệnh sau sẽ hiển thị chiều dài của đối tượng dạng đường khi được chọn:

```

Public Sub TenDT()
    Dim Obj As AcadObject
    Dim Pnt As Variant

    ThisDrawing.Utility.GetEntity Obj, Pnt, "Chon doi tuong:"
    
```

```
Debug.Print Obj.Length
End Sub
```

- Tạo đối tượng Text: mục 5.2.6 trang 232

3.2. Bài tập 2

- Lập chương trình đánh dấu các dòng chứa dữ liệu cần tìm trong bảng Excel
- Chương trình thực hiện nội dung sau:
 - Yêu cầu người dùng chọn mã vật liệu cần tìm
 - Tự động duyệt toàn bộ vùng dữ liệu hiện có và đánh dấu tất cả các dòng chứa loại vật liệu đó trong các công việc khác nhau.
- Hướng dẫn cơ bản:
 - Mở file “Phan tích vat tu.xlsx” trong thư mục dự án

	A	B	C	D	E	F
1	BẢNG PHÂN TÍCH VẬT TƯ					
2	CÔNG TRÌNH: CẢI TẠO, NÂNG CẤP ĐƯỜNG AAA					
3	HẠNG MỤC : THOÁT NƯỚC					
4						
5	STT	MÃ HIỆU	MSVT	TÊN CÔNG VIỆC HAO PHÍ VẬT TƯ	ĐƠN VỊ	KHỐI LƯỢNG
6	1	AK.98120		Làm lớp đá đệm móng tường chắn	m3	150.00
7			:A24.0026	Đá cấp phối D<=6cm	m3	180.00
8			:A24.0180	Cát vàng	m3	45.00
9	2	AE.26314		Xây gạch chỉ tường chắn bằng vữa xi măng mác 75	m3	258.00
10			:A24.0383	Gạch chỉ 6,5x10,5x22	viên	141.900.00
11			:A24.0797	Xi măng PC30	kg	27.495.06
12			:A24.0176	Cát mịn ML=1,5-2,0	m3	93.65
13			:A24.0524	Nước	lít	22.337.64
14	3	AE.11215		Xây đá học cửa cống, vữa xi măng mác 100	m3	168.00
15			:A24.0031	Đá học	m3	201.60
16			:A24.0010	Đá 4x6	m3	9.58
17			:A24.0797	Xi măng PC30	kg	27.168.42
18			:A24.0180	Cát vàng	m3	76.91
19			:A24.0524	Nước	lít	18.345.60
20	4	AK.98120		Làm lớp đá dăm đệm móng cống	m3	213.36
21			:A24.0026	Đá cấp phối D<=6cm	m3	256.03
22			:A24.0180	Cát vàng	m3	64.01
23	5	AE.26314		Xây gạch chỉ tường rãnh bằng vữa xi măng mác 75	m3	547.06
24			:A24.0383	Gạch chỉ 6,5x10,5x22	viên	300.880.80
25			:A24.0797	Xi măng PC30	kg	58.299.75
26			:A24.0176	Cát mịn ML=1,5-2,0	m3	198.56

- Khởi động VBA IDE trong Excel.
- Tạo ra Module1 để viết mã lệnh.

- Dùng hàm InputBox của đối tượng Application để nhập dữ liệu, lưu ý là chỉ chọn 1 Cell chứa mã số tập tư của loại vật liệu cần đánh dấu (trong hình dưới là ô C11, loại vật liệu là Xi măng PC30)

STT	MÃ HIỆU	MSVT	MÔ TẢ	ĐƠN VỊ	KHỐI LƯỢNG
1	AK.98120		Làm lớp đá đệm móng tường chắn	m3	150.00
		:A24.0026	Đá cấp phối D<=6cm	m3	180.00
		:A24.0180	Cát vàng	m3	45.00
2	AE.26314		Xây gạch chỉ tường chắn bằng vữa xi măng mác 75	m3	258.00
		:A24.0000	Gạch chỉ 6,5x10,5x22	viên	141,900.00
		:A24.0797	Xi măng PC30	kg	27,495.06
		:A24.0176	Cát mịn ML=1,5-2,0	m3	93.65
		:A24.0524	Nước	lít	22,337.64
3	AE.11215		Xây đá bêtông vữa, công vữa xi măng mác 100	m3	168.00

- Viết mã lệnh để đánh dấu các dòng có loại vật liệu được chọn

STT	MÃ HIỆU	MSVT	TÊN CÔNG VIỆC HAO PHÍ VẬT TƯ	ĐƠN VỊ	KHỐI LƯỢNG
BẢNG PHÂN TÍCH VẬT TƯ					
CÔNG TRÌNH: CẢI TẠO, NÂNG CẤP ĐƯỜNG AAA					
HẠNG MỤC : THOÁT NƯỚC					
1					
2					
3					
4					
5					
6	1	AK.98120	Làm lớp đá đệm móng tường chắn	m3	150.00
7		:A24.0026	Đá cấp phối D<=6cm	m3	180.00
8		:A24.0180	Cát vàng	m3	45.00
9	2	AE.26314	Xây gạch chỉ tường chắn bằng vữa xi măng mác 75	m3	258.00
10		:A24.0383	Gạch chỉ 6,5x10,5x22	viên	141,900.00
11		:A24.0797	Xi măng PC30	kg	27,495.06
12		:A24.0176	Cát mịn ML=1,5-2,0	m3	93.65
13		:A24.0524	Nước	lít	22,337.64
14	3	AE.11215	Xây đá hộc cửa cống, vữa xi măng mác 100	m3	168.00
15		:A24.0031	Đá hộc	m3	201.60
16		:A24.0010	Đá 4x6	m3	9.58
17		:A24.0797	Xi măng PC30	kg	27,168.42
18		:A24.0180	Cát vàng	m3	76.91
19		:A24.0524	Nước	lít	18,345.60
20	4	AK.98120	Làm lớp đá dăm đệm móng cống	m3	213.36
21		:A24.0026	Đá cấp phối D<=6cm	m3	256.03
22		:A24.0180	Cát vàng	m3	64.01
23	5	AE.26314	Xây gạch chỉ tường rãnh bằng vữa xi măng mác 75	m3	547.06
24		:A24.0383	Gạch chỉ 6,5x10,5x22	viên	300,880.80
25		:A24.0797	Xi măng PC30	kg	58,299.75
26		:A24.0176	Cát mịn ML=1,5-2,0	m3	198.56
27		:A24.0524	Nước	lít	47,364.11

- Sử dụng UsedRange để xác định vùng dữ liệu tự động hoặc yêu cầu người dùng nhập vùng dữ liệu bằng hàm Application.InputBox
- Mã lệnh đánh dấu 1 dòng:

```
Public Sub Test()
    Dim R As Range
    Set R = Application.InputBox("Chọn mã số vật tư (MSVT)", Type:=8)
    Dim n As Long
    n = R.Row ' Số thứ tự dòng của Cell được chọn
    ThisWorkbook.ActiveSheet.Rows(n).Interior.Color = RGB(255, 0, 0)
End Sub
```